

Chapter 2

Calculus

In this chapter we explore calculus with the help of *Maple*. Much of what is seen in this chapter should be familiar (or at least recognized) from first-year study. We aim to revise this material, as well as visualize it in ways that are, one hopes, easily accessible and in addition provide new insight even to the more capable reader. We also attempt to introduce some newer (or, at least, less familiar) material. In all cases here the aim is to use *Maple* to complement and improve our own calculus skills; the goal is one of a human/machine collaboration, not that of an electronic replacement for calculus skills.


2.1 Revision and Introduction

In this section we introduce the *Maple* commands best suited for studying and performing calculus. In addition we recall key concepts from typical first-year calculus courses. It is, however, expected that the reader is familiar with the underlying concepts and is able to perform such first-year calculus including (but not limited to) differentiation and integration of single variable functions, evaluation of limits, and curve sketching, among others. The reader is encouraged to review his or her favorite (or most readily available) calculus text.


2.1.1 Plotting

In much first-year calculus the ability to be able to visualize the functions and concepts being studied is quite valuable, but usually not readily available. Indeed in almost anything involving calculus visualization is a powerful tool. So we begin this calculus chapter by looking at how to have *Maple* plot functions.

Briefly in Section 1.1.2 we saw a plot of a cubic. The *Maple* command to plot a function is, unsurprisingly, **plot**. The most basic use of the **plot** function is to simply give it an expression involving a single variable, usually x but any valid *Maple* variable name will work just as well.




Maple automatically assigns the horizontal axis to be the axis for the independent variable (x in the previous example) and labels it as such. The vertical axis is unnamed, but depicts—as it always does—the values of the function (or the dependent variable). Also, unless we tell it otherwise, the **plot** command will plot over the horizontal interval $(-10, 10)$. If we wish to plot over a different interval, we must provide a second argument.



Notice here that our horizontal axis is now t , where it was x for the parabola example. Notice the vertical axis in these two examples. The values of the vertical axis automatically adjust to suit the function we are trying to plot. For the parabola the vertical axis was between 0 and 100, corresponding to the values the parabola would

have for $-10 \leq x \leq 10$, and the sine curve used vertical range $(-1, 1)$ just as we would have hoped it did. It should be interesting to note that the actual screen space taken up by the two plots is the same in both cases, showing that the vertical scale is different in both cases. In fact, the vertical scale and horizontal scale may be different even in the same plot, as is the case in both of these examples.

If we have a function to plot—instead of an expression—there is another way we may ask for a plot. Note above that although $\sin$ is definitely a function, $\sin(t)$ is actually a *Maple* expression. The former will take an input and produce an output, but the latter will not take any input at all. If we have a function or procedure we wish to plot, we may omit the input from the function/procedure, and omit the variable name from the second input parameter. For example, to plot the $\tan$ function between its asymptotes we can use the following.




This is equivalent to the command

$$\text{plot}\left(\tan(t), t = -\frac{\text{Pi}}{2} \dots \frac{\text{Pi}}{2}\right)$$

Unfortunately the plot we see certainly doesn't look like the $\tan$ function that we all know and love. If we have a look at the first plot we produced, we should notice the scale of the vertical axis is exceptionally large, and that we have two seemingly vertical lines at the end of the graph. If we were to somehow limit the vertical interval (and thus the vertical scale), perhaps we'd get a better picture.

As it happens, we may do just this with a third argument to the plot function. This third argument must be a range. It may be in the form of an interval $(a..b)$ or may be assigned to a variable ($\text{var} = a..b$). In the latter case, the vertical axis is labeled with the variable name. In either case, however, the vertical axis range must be present after the horizontal axis range. For example, to refine our plot of the $\tan$ function we might do the following.


```
> plot(tan, -Pi/2 .. Pi/2, -5..5)
```



We now have a much more familiar plot. To better understand why this helped, we need to know how *Maple* plots a function.

When *Maple* plots a function it evaluates that function at various points along the horizontal interval and fits a curve to the sampled points. We see this sampling behavior by asking *Maple* to plot points, instead of a line, as follows.

```
> plot(sin, 0..2*Pi, style = point)
```




By default, *Maple* uses 50 sample points, but the number of points may be specified by the **numpoints** input parameter to the **plot** function.

Usually the sample points are evenly spaced along the horizontal range, however, *Maple* is rather clever and if it detects that the values of the function are changing too quickly between sample points, it will sample the function at an extra point between them to try to obtain more and better information with which to plot the function. We observe this happening at the peaks of the curve plotted above.

This extra sampling—known as subdivision—can, by default, happen up to 6 times between any two sample points and so one should be aware that *Maple* could potentially end up evaluating a function at as many as 6 times the number of sampling points requested. This behavior is, of course, able to be controlled and modified using input parameters (see **?plot/options**).

Now, if we have a look at the tan plot in point mode, we see two extreme points far to the top and bottom of the graph, with the remainder of the sampled points on or very near the x -axis.



Of course, with the vertical range so large, the scale is such that the points seemingly on the x -axis could have values varying anywhere between $\pm 1,000,000$ or more and we wouldn't be able to tell the difference.


Without explicitly specifying the vertical range, *Maple* adjusts it accordingly between the largest and smallest sampled values. It should be clear then that by specifying the vertical range we see only the plot generated from the sampled points inside that range. It is worth noting that even with a specified vertical range *Maple* still samples all the same points as it would have without the specified range.

2.1.2 Multiple Plots

An interesting example comes to us from Borwein and Devlin [5]. Suppose we have two expressions, $y - y^2$ and $-y^2 \log(y)$, and wish to know (and eventually prove) which (if either) is always larger when $y \in (0, 1)$. A good first step would be to plot the two

expressions over the unit interval, to see if the curves cross each other. However, up until now we have only plotted single expressions. Two separate plots are of limited (if any) use to us. We need a way to plot the two expressions on the same pair of axes. This is made possible in one of two ways.

The first method is far and away the simplest. *Maple's* **plot** command will happily plot a list of expressions (or functions). In place of a single expression we simply provide a list, and any parameters that modify the plots must also be provided in a list. For example, to plot the above two functions, with the first one being colored the usual red, and the second colored blue (so we may identify which is which) we would enter the following.




For the second method, let us now consider a modified version of our example. This time we compare $y^2 - y^4$ and $-y^2 \log(y)$ (also from [5]). First we will make a simple observation: up until now, any valid *Maple* expression could be assigned to a variable name. It should, then, be a natural question to ask whether the same can be done with the plot function. The answer is that yes it can, although the logistics of such an assignment are probably not obvious. Let's try this.


```
> plot1 := plot(y^2 - y^4, y = 0..1, color = red);
   plot2 := plot(-y^2 * log(y), y = 0..1, color = blue)
                                plot1 := PLOT(...)
                                plot2 := PLOT(...)
```

Maple stores what is known as a plot structure in the variable. If we wish to see the actual plot, then we may either just ask *Maple* for the contents of the variable in the usual way, or we may use the **display** function from the **plots** package.


```
> plot1
```



```
> plots[display](plot2)
```



This is all well and good, but it doesn't really help us show multiple plots on the same axes, at least not in any different way from the method we have already used. The key to this lies in the **display** function, whose entire purpose is not so much the display of single stored plots (which, as we have seen can be displayed easily without the use of this function), but multiple plots. Just as the `plot` function will accept a list of expressions and plot them on the same axes, so will **display** accept a list (or set) of plot structures, and display them all on the same axes.



The use of **display** for the above example above may seem unnecessarily long and complicated, when we could more easily use just a single **plot** as we did in the first example. Such an observation is quite well founded. However, there are situations where the easier method is either impossible, or impractical to use. It is these cases where **display** really shines.


To illustrate this utility, we show the previous example and a plot of its sample points at the same time. In fact, we use both of the above techniques at the same time. We use a single **plot** command to produce the two curves, then a second single **plot** command to produce the two curves as points (using the `style = point` option).

It is possible to produce the same result with a single **plot** command, but is long and difficult to read, whereas the following commands are quite descriptive, and easy to follow.

```
> f := y^2 - y^4; g := -y^2 · log(y)
      f := y^2 - y^4
      g := -y^2 · log(y)

> curves := plot([f, g], y = 0..1, color = [red, blue]) :
  points := plot([f, g], y = 0..1, color = [red, blue], style = point) :
```

We now combine these two plots using the **display** function, and have the plot we wished to see. We use this technique again later in the book when demonstrating sequences whose points lie on continuous functions in Section 2.1.3.



Display is also a very valuable tool when a function plot is difficult and time consuming and we want to reuse it quickly and with certainty that it is what we wanted to draw.

2.1.3 Limits

Calculus, ultimately, all comes down to limits. So it is with limits that we begin our exploration of calculus proper. We have already seen and used very briefly in the previous chapter, the *Maple* **limit** command. We look at it in more detail here, and recall quickly the math behind limits.

We may take a limit of a sequence (of the infinite variety) or of a function. The intuitive (and mathematically imprecise) notion of a limit is a value that we may approach as closely as we could ever wish, just by traveling sufficiently far along the sequence or function.

We start with sequences. Recall that the limit, L say, of some sequence

$$\{x_n\}_{n=1}^{\infty} = x_1, x_2, x_3, \dots$$

written

$$\lim_{n \rightarrow \infty} x_n = L$$

is a number such that for every $\epsilon > 0$ we can find a natural number N so that whenever we have any other number $n \geq N$ it will be the case that $|x_n - L| \leq \epsilon$.

The sequence

$$\left\{ \frac{1}{k} \right\}_{k=1}^{\infty} = 1, \frac{1}{2}, \frac{1}{3}, \dots$$

should be familiar and has, of course, limit 0. We ask *Maple* to verify this. The *Maple* **limit** command, just like **sum** and **prod** has both an inert and active form, with the inert form being the one with the capital “L”.


```
> f := k -> 1/k; Limit(f(k), k = infinity) = limit(1/k, k = infinity)

f := k -> 1/k
lim_{k -> infinity} f(k) = 0
```

As usual, the **value** command will perform the active calculation on the inert form.

We may also, if we wish a more visual clarification of the convergence, plot this. We could simply just plot the continuous function $1/k$ to see the convergence (using the fact that if the function converges, then the sequence evaluated only at integer points also converges). Instead, however, we use *Maple*’s point plot option and see only the points of the sequence in our visualization.

To do this we construct a sequence (*Maple* sequence, that is) of 2-element lists $[x, y]$ each of which represent a point in the Cartesian plane. Because we are plotting a sequence, we choose the x -axis to be our index, and the y -axis to be the sequence element. As such the points are $[k, 1/k]$. We look at the first 100 points. This sequence of lists is then put into a containing list, so the **plot** function does not get confused.



The convergence is visually pretty clear. It is worth stressing at this point, however, that these plots give an *indication* of convergence, not a proof of convergence. There is always the possibility that the sequence does something odd after the interval we have plotted. So we must still perform regular mathematics to verify the limits, or at the very least ask *Maple* to evaluate the limit.

Recall now that an infinite sum is defined to be a limit of its partial sums. Mathematically, that is,

$$\sum_{k=1}^{\infty} f(k) = \lim_{N \rightarrow \infty} \sum_{k=1}^N f(k)$$

We have already used *Maple*'s **sum** command to calculate infinite sums for us in Section 1.1.5, but we do this again now, and demonstrate the limit property. Let us use the series $1/k^2$ again, which we know from the previous section converges to $\pi^2/6$.

```

> sum(1/k^2, k = 1..N);
Limit(% , N = infinity) = limit(% , N = infinity)
                        -Psi(1, N + 1) + 1/6 pi^2
                        lim_{N -> infinity} (-Psi(1, N + 1) + 1/6 pi^2) = 1/6 pi^2

```

We have not seen the Ψ function before, although we may ask *Maple* about it by using the command **?Psi**. We should expect, due to the algebra of limits, that $\lim_{N \rightarrow \infty} -\Psi(1, N + 1) = 0$. It is always good to cross check answers we are unfamiliar with, we therefore do so.

```


> limit(-Psi(1, N + 1))
0

> plot(-Psi(1, N + 1), N = 1..100)

```

The convergence in the picture seems pretty clear, and combined with the results *Maple* gave us for the infinite sum as well as the limit of the partial sums, and the limit of the Ψ function—not to mention the numeric approximations we saw in the previous section—we may be quite confident of the validity of the answer.

Now let us look at limits of continuous functions. For this purpose we consider the function $f(x) = 1/(x^3 - 2x^2)$. Our first impulse should be to plot the function to see what it looks like, but doing so produces something that is not so useful.



We could attempt to use trial and error to find a good interval to plot over, but instead we perform some calculus. First note that the denominator is equal to $x^2(x-2)$, which tells us that the function is not defined at $x = 0$ or $x = 2$ and that we should probably expect vertical asymptotes at those points. It should also be clear, using the algebra of limits, that

$$\lim_{x \rightarrow a} \frac{1}{x^3 - 2x^2} = \lim_{x \rightarrow a} \frac{\frac{1}{x^3}}{1 - \frac{2}{x}} = \frac{\lim_{x \rightarrow a} \frac{1}{x^3}}{1 - \lim_{x \rightarrow a} \frac{2}{x}}$$


and so for $a = \pm\infty$ the limit will be 0.

The limits at $a = 0$ and $a = 2$ are only a little bit trickier to work out. We again look at the factored denominator $x^2(x-2)$ and observe that

$$\lim_{x \rightarrow 0} x^2(x-2) = 0 \quad \text{and} \quad \lim_{x \rightarrow 2} x^2(x-2) = 0$$

Furthermore we can see that x^2 will always be ≥ 0 , and that for all points near $a = 0$ it is the case that $x - 2 < 0$ so the denominator near $a = 0$ must always be negative. We conclude therefore that the limit at $a = 0$ is $-\infty$. Finally observe that for $x > 2$ we have $x - 2 > 0$ and that for $x < 2$ we have $x - 2 < 0$ which tells us that $f(x) \rightarrow -\infty$ as $x \rightarrow 2^-$ and that $f(x) \rightarrow \infty$ as $x \rightarrow 2^+$.

We now have plenty of information for us to choose appropriate plotting ranges. In order to put the undefined points (with the vertical asymptotes) evenly spread across the x -axis we plot across the interval $x \in (-2, 4)$. However, because we know that the function has vertical asymptotes, we should limit the y -axis range somewhat. A quick substitution of $x = 1$ into the function shows that the midpoint between the vertical asymptotes attains the value -1 , and so we take a reasonable guess that $y \in (-5, 5)$ will suffice.



The *discont = true* argument to the plot command simply tells **plot** that we are plotting a discontinuous function allowing for a better plotting. Without that argument, **plot** assumes it is plotting a continuous function and ends up putting vertical lines at points of discontinuity, or at asymptotes.

We now verify the limits with *Maple*.

`> f :=  $\frac{1}{x^3 - 2x^2}$ ;`
`limit(f, x = -infinity), limit(f, x = 0), limit(f, x = 2), limit(f, x = infinity)`

$$f := \frac{1}{x^3 - 2x^2}$$

$$0, -\infty, \text{undefined}, 0$$

The undefined limit at $a = 2$ should be completely unsurprising thanks to the plot we performed above. The limit from below and the limit from above are not the same. Recall that $\lim_{x \rightarrow a^-} = \lim_{x \rightarrow a^+} = L$ if and only if $\lim_{x \rightarrow a} = L$. Fortunately for us *Maple* can handle single directional limits by allowing the **limit** command to take a third input variable for this purpose, which may be one of *left* or *right*. As might be expected, the left limit at a is when $x \rightarrow a^-$ and the right limit at a is when $x \rightarrow a^+$.

`> Limit(f, x = 2, left) = limit(f, x = 2, left);`
`Limit(f, x = 2, right) = limit(f, x = 2, right);`

$$\lim_{x \rightarrow 2^-} \frac{1}{x^3 - 2x^2} = -\infty$$

$$\lim_{x \rightarrow 2^+} \frac{1}{x^3 - 2x^2} = \infty$$


We now return to sequence limits, and a cautionary example. *Maple*'s **limit** function calculates real- or complex-valued (function) limits. When we used it to calculate sequence limits above, what we were really doing was evaluating the real-valued limit at infinity of the functions in question, and using the theorem which states that if

$f(x) \rightarrow L$ as $x \rightarrow \infty$ exists for a real-valued function f , then the sequence $\{f(n)\}_{n \in \mathbb{N}}$ converges to the same limit as $n \rightarrow \infty$.


However, caution is advised. This relationship between limits of functions and limits of sequences does not always work the other way around. Consider the function

$$f(x) = x \sin(\pi \cdot x) + 2 \tanh(x - 5)$$

and the corresponding sequence $\{f(n)\}_{n \in \mathbb{N}}$. If we evaluate limits or plot the function we might very well be tempted to conclude that the sequence does not converge.




We clearly see a function with no limit at infinity. However, plotting only the sequence points shows an entirely different picture.¹



¹ The astute reader will notice that this plot is not square, as the others are. This is because, due to the information displayed in the plot and aesthetic reasons, the author manually shrunk the height (using the mouse and a drag action).

Our sequence has what looks very much like a limit. Applying our calculus knowledge, we see that $n \sin(\pi n) = 0$ for $n \in \mathbb{N}$, and so that part of the function will not affect the sequence at all. Furthermore, because $\tanh(x) \rightarrow 1$ as $x \rightarrow \infty$ it must be the case that $2 \tanh(x - 5) \rightarrow 2$ as $x \rightarrow \infty$. In short, the sequence $\{f(n)\}_{n \in \mathbb{N}}$ ought to converge to the value 2, and the plot of the sequence points exhibited precisely this behavior.

If we plot the real-valued function and the sequence on the same axes, we can see the convergence a little better.



All that remains is to see if we may convince *Maple* to provide us with the correct answer for the sequence limit. Because we are evaluating a limit at infinity we are already evaluating a left limit and cannot possibly try to change to a right limit or a bidirectional limit. A good attempt would be to use the **assuming** keyword.

```
> limit(f(n), n = infinity) assuming n :: posint
undefined
```

Unfortunately, as it turns out, due to n being a dummy variable inside the **limit** function, the assumptions we requested for n do not quite seem to be being applied. See **?assuming** for more details. The solution is to use the **assume** command, which works in a similar way to **assuming** except that the assumptions become globally accepted for the entire worksheet (as opposed to just the current command). So we make sure to explicitly reset the n variable after performing our calculation.

```
> assume(n :: posint); limit(f(n), n = infinity); n := 'n' :
2
```

To reiterate the key points here, the lack of a limit of a real valued function does not imply the lack of a limit of a sequence of evaluations of that function, and—more important—one must always keep one's wits about one when using a CAS (of course this is true when reading a book or taking a bus too).

2.1.4 Differentiation

Differentiation is, fundamentally, all about calculating rates of change. Recall that the derivative of a function, $f(x)$ say, at any point a is the slope of the tangent line to f at the point a . Recall, also, that the tangent at a is defined to be the line through a with slope equal to the limit of the slopes of lines drawn between a and points near a , as depicted in Figure 2.1. So it is that we come to the limit definition of the derivative




Fig. 2.1 Depiction of the convergence of lines to the tangent.

at a point a as

$$f'(a) = \lim_{x \rightarrow a} \frac{f(x) - f(a)}{x - a} = \lim_{h \rightarrow 0} \frac{f(a + h) - f(a)}{h}$$

Let us explore this a little before we introduce *Maple*'s native differentiation commands. We start with a parabola $f(x) = x^2$ and so, of course, we know that $f'(x) = 2x$ meaning that the tangent to the parabola at any point a has slope $2a$. We write a small procedure to output the limit and its value.

```

> d := proc(f :: procedure, a)
    Limit( $\frac{f(a + h) - f(a)}{h}$ , h = 0) = limit( $\frac{f(a + h) - f(a)}{h}$ , h = 0)
end :

```

$$\left[\begin{array}{l} > d(x \rightarrow x^2, 1); d(x \rightarrow x^2, 2); d(x \rightarrow x^2, x) \\ & \lim_{h \rightarrow 0} \frac{(1+h)^2 - 1}{h} = 2 \\ & \lim_{h \rightarrow 2} \frac{(2+h)^2 - 4}{h} = 4 \\ & \lim_{h \rightarrow 0} \frac{(x+h)^2 - x^2}{h} = 2x \end{array} \right.$$

If we look a little closer at the final limit, we should see that

$$(x+h)^2 - x^2 = x^2 - 2hx - h^2 - x^2 = h(2x - h)$$

and so the limit then becomes


$$\lim_{h \rightarrow 0} \frac{h(2x - h)}{h} = \lim_{h \rightarrow 0} (2x - h) = 2x$$

thus verifying both *Maple's* limit calculation and our regular differentiation technique (for the parabola, at least).

Turning our eye now to trigonometric functions, we choose the sin function and the point $\pi/2$. This produces a limit that is a little trickier to work out on paper.

$$\left[\begin{array}{l} > d\left(\sin, \frac{\pi}{2}\right); \\ & \lim_{h \rightarrow 0} \frac{\cos(h) - 1}{h} = 0 \end{array} \right.$$

As ever, our instinct should be to plot the function. However, to be on the safe side, we ask for a lot of sample points, to try to get a good idea of the behavior near the $h = 0$ point which we know is undefined.



2.1.5 Integration

Integration grows out of the problem of calculating area underneath a curve, although its applications are far more wide and varied than that. Recall that a definite integral of a continuous function f between two points a and b may be approximated by a limit of rectangles.

$$\int_a^b f(x) dx = \lim_{n \rightarrow \infty} \left(\Delta x \sum_{k=1}^n f(a + k\Delta x) \right) \quad \text{where } \Delta x := \frac{b-a}{n}$$

In fact, the approximation can be made from rectangles coming from an arbitrary partitioning of the interval (a, b) with the heights of the rectangles being taken from arbitrary points within each of the partition elements. See [12] or any elementary calculus text for more details.

```

> integral := proc(f :: procedure, r :: range) local Delta, a, b;
    a, b := lhs(r), rhs(r); Delta :=  $\frac{b-a}{n}$ ;
    Limit(Delta · Sum(f(a + k · Delta), k = 1..n), n = infinity) =
    limit(Delta · sum(f(a + k · Delta), k = 1..n), n = infinity)
end :

> integral(x → x2, 0..2); integral(sin, 0..Pi); integral(sin, 0..2 · Pi)


$$\lim_{n \rightarrow \infty} \left( \frac{2 \left( \sum_{k=1}^n \frac{4k^2}{n^2} \right)}{n} \right) = \frac{8}{3}$$



$$\lim_{n \rightarrow \infty} \left( \frac{\pi \left( \sum_{k=1}^n \sin \left( \frac{k\pi}{n} \right) \right)}{n} \right) = 2$$



$$\lim_{n \rightarrow \infty} \left( \frac{2\pi \left( \sum_{k=1}^n \sin \left( \frac{2k\pi}{n} \right) \right)}{n} \right) = 0$$


```

Once this limit is established, then we are presented with the fundamental theorem of calculus which states that

$$\int_a^b f(x) dx = F(b) - F(a) \quad \text{where } F' = f$$

and nicely links differentiation and definite integrals. We also, by convention, denote the indefinite integral

$$\int f(x) dx = F(x) \Leftrightarrow F' = f$$

We may use this to check the limits found above. First we have $x^3/3$ as an antiderivative of x^2 , and evaluating

$$\frac{2^3}{3} - \frac{0}{3} = \frac{8}{3}$$

verifies the integral. Similarly we have $-\cos$ as an antiderivative of $\sin$ leading us to $-\cos(\pi) - (-\cos(0)) = 1 + 1 = 2$ and $(-\cos(2\pi) - (-\cos(0))) = -1 + 1 = 0$. In fact, the final integral can be verified by the symmetric nature of the cosine graph between 0 and 2π .

Again, as with differentiation, we need not perform limit calculations within *Maple* if we wish to calculate an integral, as there is a handy function named **int** (and its inert form **Int**). The **int** function can handle both definite and indefinite integrals and takes an expression as its first argument, and a range as its second argument in the form $var = a..b$ where var is the variable over which to integrate. In the case of an indefinite integral, the second parameter is just var with no range.

```
[ > Int(x^2, x = 1..3) = int(x^2, x = 1..3);
    Int(sin(x), x) = int(sin(x), x)

                                 $\int_1^3 x^2 dx = \frac{26}{3}$ 

                                 $\int \sin(x) dx = -\cos(x)$ 
```

The **int** command will also accept function (as opposed to expression) for its first parameter. This works in precisely the same way as the **plot** function does, also requiring the omission of $var =$ from the second parameter. However, it can only do this for definite integrals, and the inert form does not behave very well this way.

```
[ > Int(sin, 0..Pi) = int(sin, 0..Pi)
                                Int(sin, 0..pi) = 2

> value(lhs(%))
                                2
```

To close this section, we mention that the **Student** and **IntegrationTools** packages each contain many useful tools to help manipulate integrals, for instance to extract the integrand quickly without learning a lot about operands in *Maple* or resorting to cutting and pasting. The reader is, as always, actively encouraged to utilize *Maple*'s help files to learn more.

2.2 Univariate Calculus

2.2.1 Optimization

Suppose we wish to calculate the longest ladder that we may carry around a corner with one corridor 2 meters in width, and the other 1 meter in width. This is an example of an Optimization problem. We may use calculus (specifically, differentiation) to solve problems along these lines.


Given any angle θ we know that

$$x = \frac{1}{\cos(\theta)} \quad \text{and} \quad y = \frac{2}{\sin(\theta)}$$

Therefore, the length L of a ladder that touches the corner, and the opposite walls of each corridor at an angle of θ to the corner, is given by the formula


$$L := \frac{1}{\cos(\theta)} + \frac{2}{\sin(\theta)}$$

We can see straight away that $0 < \theta < \pi/2$ and that $1/\cos(\theta) \rightarrow \infty$ as $\theta \rightarrow (\pi/2)^-$ as well as $2/\sin(\theta) \rightarrow \infty$ as $\theta \rightarrow 0^+$ which tells us that our L function will tend toward infinity at its endpoints. We also know that $L > 0$ for all values of θ , just as we would expect for the length of a ladder. We plot the function, being sure to limit the y -axis.



What is interesting here is that the function is concave up and has no maximum values, although it does have a minimum value. What's going on here? Well, our function L , as we know, calculates the length of a ladder at an angle of θ that touches both the far walls of the two corridors as well as the corner. Such a ladder is the longest possible ladder that can rest at that particular angle. However, if a ladder is to be carried around the corner, then it must go through all angles from $\theta = 0$ to $\theta = \pi/2$ and not become stuck. What we are, in essence, looking for is the shortest of all such longest ladders, and hence a minimum of the function L .

With a little trial and error, we may find the following plot and see that the minimum value lies somewhere in the range $0.5 \leq \theta \leq 1$. We can, in fact, do a little better with these bounds. Looking at the plot (below) we can clearly see the the minimum lies to the right of the midpoint of the θ -axis. Even though it is not labelled, we know this midpoint must be $\pi/4$ because our plot is for theta between 0 and $\pi/2$. Our minimum, therefore, must lie in the range $\pi/4 \leq \theta \leq 1$.



We could, if we wished “zoom in” by plotting the region $\pi/4 \leq \theta \leq 1$ and $4 \leq L \leq 5$ in the hopes of obtaining better bounds. We could even continue along these lines for some time. There is little point in doing so, at least in this case. Instead we proceed to find the minimum symbolically. The minimum is a turning point, and so will have a derivative of 0. We therefore solve $L'(\theta) = 0$.

```
> D(L)(theta); solve(% = 0)
```

$$\frac{\sin(\theta)}{\cos(\theta)^2} - \frac{2 \cos(\theta)}{\sin(\theta)^2}$$

$$\arctan(2^{1/3}), -\arctan\left(\frac{1}{2} 2^{1/3} - \frac{1}{2} I \sqrt{3} 2^{1/3}\right), -\arctan\left(\frac{1}{2} 2^{1/3} + \frac{1}{2} I \sqrt{3} 2^{1/3}\right)$$

Well, that’s a bit of a mess. *Maple* seems to have given us three solutions, two of which appear to be complex. Nonetheless, there is clearly a real solution there, and evaluating it numerically shows it to be in the range we were expecting.

```
> evalf(arctan(2^(1/3))); L = L(arctan(2^(1/3))); evalf(%)
```

$$0.8999083481$$

$$L = \sqrt{2^{2/3} + 1} + 2^{2/3} \sqrt{2^{2/3} + 1}$$

$$L = 4.161938184$$

And so there we have it. A ladder of approximately 4.16 meters in length is the longest we may carry around the corner. We can see quite well from the plots above that this is clearly a local minimum, however, a quick second derivative test won’t hurt.

```
> D(2)(L)(arctan(2^(1/3))); is(% > 0); evalf(%%)
```

$$3 2^{2/3} \sqrt{2^{2/3} + 1} + 3 \sqrt{2^{2/3} + 1}$$

$$true$$

$$12.48581455$$

2.2.2 Integral Evaluation

Integral evaluation can, at times, be quite tricky. A tool such as *Maple* can indeed be an asset, however, even it may be unable to perform certain integrals symbolically. For example, suppose we ask *Maple* to integrate xe^{x^3} between 0 and 1.

```
> int(x · exp(x3), x = 0..1)
```

$$\int_0^1 xe^{x^3} dx$$

Notice here that we definitely used the active form of the integral command, and yet *Maple* still returned only the integral back again. It seems that *Maple* cannot provide a better symbolic answer than the integral itself. The use of hand-methods should prove equally frustrating. We note that if the integrand were xe^{x^2} , then the task would be trifling easy. At times like this, all one can really do is ask for a numerical answer

```
> evalf[50](int(x · exp(x3), x = 0..1)); identify(%);
0.78119703110865591510743281434829950577669739096218
0.78119703110865591510743281434829950577669739096218
```

The use of the **identify** is a request to *Maple* to take a good guess at a likely symbolic representation of the given decimal number. In this case **identify** simply gives us back the decimal number, meaning that it could not find any likely symbolic representation. Similarly, neither Sloane's On-line Encyclopedia of Integer Sequences², nor the Inverse Symbolic Calculator³ turn up anything. It looks like we're stuck with just a numerical approximation of the integral.

Suppose we wish to evaluate the integral

$$\int_0^\pi \frac{x \sin(x)}{1 + \cos^2(x)} dx$$

Well, our first attempt should be to see what *Maple* thinks.

```
> simplify(int(x · sin(x) / (1 + cos(x)2), x = 0..Pi))
```

$$\frac{1}{4}\pi^2 - \operatorname{dilog}\left(\frac{\sqrt{2}-1-I}{\sqrt{2}-1}\right) + \operatorname{dilog}\left(\frac{1-I+\sqrt{2}}{1+\sqrt{2}}\right) - \operatorname{dilog}\left(\frac{1+I+\sqrt{2}}{1+\sqrt{2}}\right) + \operatorname{dilog}\left(\frac{\sqrt{2}-1+I}{\sqrt{2}-1}\right) + I\pi \ln(1+\sqrt{2})$$

That's not really very useful. Let's try to evaluate that mess as a decimal number.

```
> evalf[50](%)
2.4674011002723396547086227499690377838284248518102 - 3. 10-49 I
```

Hmmm, that's a complex number even if the complex part is infinitesimal. Let's try again, and see if asking *Maple* to evaluate the integral numerically works any better. We even throw in an **identify** for good measure.

² <http://oeis.org/> at the time of writing

³ <http://isc.carma.newcastle.edu.au/> at the time of writing

```

> int( (x * sin(x)) / (1 + cos(x)^2), x = (0)..Pi ); identify(%)
2.467401100
1/4 pi^2

```

Numerically evaluating the integral, and asking *Maple* to guess a symbolic value, gives us that the integral is very probably equal to $\frac{1}{4}\pi^2$. At this point we would start looking for a more formal proof. Fortunately, in this case, it can be shown (see Exercise 5) that

$$\int_0^\pi x f(\sin(x)) dx = \frac{\pi}{2} \int_0^\pi f(\sin(x)) dx$$

and it should be pretty clear that

$$\frac{\sin(x)}{1 + \cos^2(x)} = \frac{\sin(x)}{2 - \sin^2(x)} = f(\sin(x)) \text{ where } f = \frac{x}{2 - x^2}$$

which leaves us with the rather simpler integral

$$\int_0^\pi \frac{x \sin(x)}{1 + \cos^2(x)} dx = \frac{\pi}{2} \int_0^\pi \frac{\sin(x)}{1 + \cos^2(x)} dx$$

As it happens, this new integral is easier for *Maple* to handle.

```

> Pi/2 * int( sin(x) / (1 + cos^2(x)), x = 0..Pi )
1/4 pi^2

```

Let us perform another integral. This time we evaluate

$$\int_0^\infty \frac{x^2}{\sqrt{e^x - 1}} dx$$

As a first attempt, we see what *Maple* makes of the integral.

```

> int( x^2 / (sqrt(exp(x) - 1)), x = 0..infinity )
int_0^infinity x^2 / sqrt(e^x - 1) dx

```

Unfortunately, *Maple* apparently cannot evaluate this integral. Our next course of action is to attempt numeric evaluation.

```

> evalf(%); identify(%)
16.37297620
9/2 * 2^(3/5) * sqrt(3) * zeta(5)^9

```

We now have a possible value for the integral. Remember, however, that the **identify** function gives a *probable* symbolic expression but not a certain one. We return shortly to the question of how reliable this symbolic guess is, but it should be clear that more work is needed to be performed in order to verify the identity with any certainty.

We have a closer look at the integral. Looking at the denominator, $\sqrt{e^x - 1}$, we see that a substitution of $x = \log(u)$ will change the denominator to $\sqrt{u - 1}$ thus

eliminating the exponent term. Performing this substitution, we see that

$$\frac{dx}{du} = \frac{1}{u} \Rightarrow dx = \frac{du}{u}$$

It is also the case that $\log(0) = 1$ and $\log(x) \rightarrow \infty$ as $x \rightarrow \infty$ so that

$$\int_0^\infty \frac{x^2}{\sqrt{e^x - 1}} dx = \int_1^\infty \frac{\log(u)^2}{u\sqrt{u-1}} du$$

Turning again to *Maple* with this new integral we find altogether better success than we did previously.

$$\left[\begin{array}{l} > \text{int}\left(\frac{\log(u)^2}{u \cdot \text{sqrt}(u-1)}, x = 1..infinity\right) \\ \\ \frac{1}{3} \pi^3 + 4 \pi \ln(2)^2 \end{array} \right]$$

Interestingly, however, we have an altogether different symbolic answer to the one produced by **identify**, above. It is prudent then, to perform a sanity check

$$\left[\begin{array}{l} > \text{evalf}(\%) \\ \\ 16.37297620 \end{array} \right]$$

which is precisely the same 10-digit value we obtained from our first attempt. This is absolutely expected, of course, because substitutions do not alter the value of an integral.

As an epilogue to this example, we return to the question of the symbolic answer that **identify** gave us earlier. It turns out that if we obtain even one extra digit of precision for this integral, then *Maple* is completely unable to identify the decimal approximation.

$$\left[\begin{array}{l} > \text{evalf}[11]\left(\text{int}\left(\frac{x^2}{\text{sqrt}(\exp(x)-1)}, x = 0..infinity\right)\right); \text{identify}(\%) \\ \\ 16.372976196 \\ 16.372976196 \end{array} \right]$$

Indeed, for any precision greater than 11 digits, it seems that *Maple* is unable to identify the number.

Exploring this a little more closely, we find that the two differ after, approximately, eight decimal places.

$$\left[\begin{array}{l} > \text{evalf}\left(\text{int}\left(\frac{x^2}{\text{sqrt}(\exp(x)-1)}, x = 0..infinity\right) - \frac{9}{2} \cdot 2^{3/5} \cdot \text{sqrt}(3) \cdot \text{Zeta}(5)^9\right) \\ \\ -4.10^{-8} \\ \\ > \text{evalf}[50]\left(\text{int}\left(\frac{x^2}{\text{sqrt}(\exp(x)-1)}, x = 0..infinity\right) - \frac{9}{2} \cdot 2^{3/5} \cdot \text{sqrt}(3) \cdot \text{Zeta}(5)^9\right) \\ \\ -5.7359902850345135770849920305260620836853 \cdot 10^{-8} \end{array} \right]$$

It would seem that **identify** was confused by not having enough decimal digits of the number in question with which to work.

At least two things should be readily apparent from these examples. First and foremost is that *Maple* is not a substitution for poor calculus skills (or, at least, is a poor one). Second, answers—especially from the **identify** function—should be checked from a number of avenues before being accepted. Human/machine collaboration is the name of the game here. *Maple* can be wonderful for performing tedious calculations quickly,

and is remarkably adept at performing integral calculus, but a correct substitution, or other mathematical insight on our part can mean the difference between successfully obtaining a symbolic answer, or not.

2.2.3 Symbolic Integrals

In the previous section we took pains to calculate the value of certain definite integrals, but did not always find a symbolic answer. Our attempts to find a symbolic answer, however, began and ended with issuing an **int** command. In one case we performed a substitution, but this amounted to a second (and successful, as it happened) attempt at issuing an **int** command. The reader may wonder if it is possible to have *Maple* perform such a substitution. The astute reader may realize that an integral itself is just another mathematical expression that is subject to symbolic manipulation, and may further wonder whether *Maple* can treat one as such. In fact, these two questions are related, and the answer to both is yes and involves the **IntegrationTools** package.

```
[ > with(IntegrationTools)
    [Change, Combine, Expand, Flip, GetIntegrand, GetRange, GetVariable, Parts,
      Split]
```

There are not many functions here, but what is there should look familiar. The reader should spend some time looking at the help files for this package (**?IntegrationTools**). It should be mentioned that the **IntegrationTools** functions seem to work better with inert integrals, and can be a little unpredictable with active integrals. Let us look first at the integral from the previous example.

```
[ > A := Int( (x^2 / sqrt(exp(x) - 1)), x = 0..infinity )
      A := ∫₀^∞ (x² / √(eˣ - 1)) dx
```

We know, from above, that *Maple* cannot evaluate this integral directly. We also know, from above, exactly what the substitution we need to perform is; $x = \log(u)$. It should be little surprise that the function we need to use to have *Maple* perform this substitution is the **Change** function (for a change of variable).

```
[ > Change(A, x = log(u))
      ∫₁^∞ (ln(u)² / √(u - 1)u) du
```

We could just as easily have phrased the change of variable as $u = e^x$

```
[ > Change(A, u = exp(x))
      ∫₁^∞ (ln(u)² / √(u - 1)u) du
```

Either way we have the integral that we calculated by hand in the previous section, and we know that *Maple* can compute this integral

```
[ > value(%)
      1/3 π³ + 4 π ln(2)²
```

We look now at a different example. We calculate the integral of $e^x \cos(x)$. As it happens, *Maple* can calculate this directly.

```
[ > B := Int(exp(x) · cos(x), x)
                                     B := ∫ ex cos(x) dx
[ > value(B)
                                     1/2 ex cos(x) + 1/2 ex sin(x)
```

We explore this integral a little. If we were to try this by hand, our first instinct should be to try to use integration by parts. Setting $u = e^x$ and $dv = \cos(x)$ we have that $du = e^x$ and $v = \sin(x)$. We use the **Parts** function from the **IntegrationTools** package to perform this integration by parts. Much like the **Change** function, we first specify the integral on which we wish to work. The second parameter is the value of u for the integration by parts. An optional third parameter may also be used to specify the v parameter directly, but we do not use this here. The interested reader should consult the help files.

```
[ > B = Parts(B, exp(x))
                                     ∫ ex cos(x) dx = ex sin(x) - ∫ ex sin(x) dx
```

In order to make any use of this, we need to perform the same integration by parts on this new integral. One useful feature of the **IntegrationTools** functions is that they only operate on integrals, and tend to ignore anything else. This may sometimes be inconvenient in the case of a change of variables, but in this case we exploit the feature.

```
[ > C := Parts(B, exp(x))
                                     C := ex sin(x) - ∫ ex sin(x) dx
[ > C := Parts(C, exp(x))
                                     C := ex sin(x) + ex cos(x) + ∫ -ex cos(x) dx
```

We now have the integral of $e^x \cos(x)$ again, which is what we started with, although there's a negative in this newest one. We can pull the negative outside the integrand with the **simplify** command.

```
[ > C := simplify(C)
                                     C := ex sin(x) + ex cos(x) - ∫ ex cos(x) dx
```

We now have the following identity.

```
[ > B = C
                                     ∫ ex cos(x) dx = ex sin(x) + ex cos(x) - ∫ ex cos(x) dx
```

We can see the result easily enough from this identity, however, we have *Maple* produce it for us in order to demonstrate how we may perform arithmetic on equalities. The first step is to add the integral to both sides; then we divide by 2.

```
[ > % + B
                                     2 ∫ ex cos(x) dx = ex sin(x) + ex cos(x)
```

$$\left[\begin{array}{l} > \frac{\%}{2} \\ \int e^x \cos(x) dx = \frac{1}{2} e^x \sin(x) + \frac{1}{2} e^x \cos(x) \end{array} \right.$$

2.2.4 Differential Equations

Differential equations are equations that relate a function to its derivatives. A solution to a differential equation is a function that has the required relationship with its derivatives. The simplest differential equation is $y' = y$ which has the solution $y = Ce^x$ (where C is an arbitrary constant). This should be nothing new to anybody who has studied first-year calculus.

A first-order linear differential equation can always be re-written to have the form

$$y' + P(x)y = Q(x)$$

and can be solved by use of an *integrating factor*

$$I(x) := e^{\int P(x) dx}$$

which has the property that

$$\int \left(I(x) y' + I(x) P(x) y \right) dx = I(x) y$$

It is fairly simple to verify the claimed property of the integrating factor by hand. We check it in *Maple*.

$$\left[\begin{array}{l} > IF := \exp(\text{int}(P(x), x)); \\ & IF := e^{\int P(x) dx} \\ > \text{int}(IF \cdot \text{diff}(y(x), x) + IF \cdot P(x) \cdot y(x), x) \\ & e^{\int P(x) dx} y(x) \\ > \text{diff}(IF \cdot y(x), x) \\ & e^{\int P(x) dx} \left(\frac{d}{dx} y(x) \right) + e^{\int P(x) dx} P(x) y(x) \end{array} \right.$$

Of course, we only needed to check one of these identities, the other one we get for free with the fundamental theorem of calculus.

With this identity available to us, we may solve the equation by multiplying the left-hand and right-hand sides by the integrating factor, integrating both sides, and solving for y . The solution, therefore, to the differential equation is

$$\begin{aligned} y' + P(x)y = Q(x) &\implies \int I(x)(y' + P(x)y) dx = \int I(x)Q(x) dx \\ &\implies I(x)y = \int I(x)Q(x) dx \end{aligned}$$

and so the solution is

$$y = \frac{\int I(x) Q(x) dx}{I(x)}$$

Let us now consider the linear differential equation $xy' + y = 3x^3$, which may be rewritten as $y' + x^{-1}y = 3x^2$; then we may use *Maple*

$$\left[\begin{array}{l} > P := x \rightarrow \frac{1}{x}; Q := x \rightarrow 3 \cdot x^2; IF := \exp(\text{int}(P(x), x)) \\ & P := x \rightarrow \frac{1}{x} \\ & Q := x \rightarrow 3x^2 \\ & IF := x \\ > \frac{\text{int}(IF \cdot Q(x), x) + C}{IF} \\ & \frac{\frac{3}{4}x^4 + C}{x} \end{array} \right]$$

Note, however, that for a general solution we needed to add manually the constant of integration when calculating the answer, because *Maple*'s **int** function does not include this constant.

We may cross-check this using *Maple*'s inbuilt differential equation solving function **dsolve**.

$$\left[\begin{array}{l} > \text{dsolve}(x \cdot \text{diff}(y(x), x) + y(x) = 3 \cdot x^3) \\ & y(x) = \frac{\frac{3}{4}x^4 + _C1}{x} \end{array} \right]$$

Moving on now to second-order differential equations. A *second-order linear differential equation* is a differential equation of the form

$$P(x)y'' + Q(x)y' + R(x)y = G(x)$$

and is furthermore said to be *homogeneous* if $G(x) = 0$. Finally, if the functions $P(x)$, $Q(x)$, and $R(x)$ are constant functions then the differential equation is said to have *constant coefficients*; however, if the differential equation is still to be second-order then $P(x) \neq 0$.

Homogeneous second-order linear differential equations with constant coefficients may be solved in a manner almost identical to that used to solve homogeneous second-order linear recurrence relations with constant coefficients, which we looked at in Section 1.3.3.

Given the equation $ay'' + by' + c = 0$ we construct the *characteristic equation* $at^2 + bt + c = 0$ and solve for t . There are only three possibilities for the roots r_1, r_2 of the equation. The general formula is as follows.

$$y(x) = \begin{cases} Ae^{r_1x} + Be^{r_2x} & r_1 \neq r_2 \\ Ae^{r_1x} + Bxe^{r_2x} & r_1 = r_2 \\ e^{\alpha x} (A \sin(\beta x) + B \cos(\beta x)) & r_1, r_2 = \alpha \pm i\beta \end{cases}$$

where A and B are arbitrary constants.

Let's look at some examples. We start with a differential equation that has the same characteristic equation as the Fibonacci numbers, $y'' - y' - y = 0$. Because this has characteristic equation $t^2 - t - 1$, we know from Section 1.3.2 that the roots are

$t = \frac{1}{2}(1 \pm \sqrt{5})$. As such, the solution to the differential equation should be

$$y = Ae^{x \cdot (1+\sqrt{5})/2} + Be^{x \cdot (1-\sqrt{5})/2}$$

We check this in *Maple*.

$$\left[\begin{array}{l} > y := A \cdot \exp\left(\frac{x \cdot (1 + \text{sqrt}(5))}{2}\right) + B \cdot \exp\left(\frac{x \cdot (1 - \text{sqrt}(5))}{2}\right) \\ & Ae^{\frac{1}{2}x(1+\sqrt{5})} + Be^{\frac{1}{2}x(1-\sqrt{5})} \\ > \text{diff}(y, x); \text{diff}(y, x, x) \\ & A \left(\frac{1}{2} + \frac{1}{2}\sqrt{5}\right) e^{\frac{1}{2}x(1+\sqrt{5})} + B \left(\frac{1}{2} - \frac{1}{2}\sqrt{5}\right) e^{\frac{1}{2}x(1-\sqrt{5})} \\ & A \left(\frac{1}{2} + \frac{1}{2}\sqrt{5}\right)^2 e^{\frac{1}{2}x(1+\sqrt{5})} + B \left(\frac{1}{2} - \frac{1}{2}\sqrt{5}\right)^2 e^{\frac{1}{2}x(1-\sqrt{5})} \\ > \text{simplify}(\text{diff}(y, x, x) - \text{diff}(y, x) - y) \\ & 0 \end{array} \right]$$

And asking *Maple* for the solution directly also gives the expected result (which is always nice).

$$\left[\begin{array}{l} > y := 'y'; \text{dsolve}\left(D^{(2)}(y)(x) - D(y)(x) - y(x) = 0\right) \\ & y(x) = _C1 e^{\frac{1}{2}x(1+\sqrt{5})} + _C2 e^{-\frac{1}{2}x(1+\sqrt{5})} \end{array} \right]$$

In the case of nonhomogeneous second-order linear differential equations with constant coefficients the solution is obtained by first calculating the solution to the equivalent homogeneous differential equation (essentially just pretending that $G(x) = 0$) and adding to that a *particular* solution.

To be more mathematically rigorous here, given the equation

$$ay'' + by' + cy = G(x)$$

the general solution is

$$y = y_p + y_c$$

where y_p is some (any) solution to the equation, and y_c is the solution to the complementary homogeneous equation

$$ay'' + by' + cy = 0$$

which is henceforth referred to only as the *complementary equation*.

Ordinarily in first-year courses the particular solution is obtained by a guess-and-check approach. A more systematic method named *variation of parameters* is described in [12], but tends to be trickier and slower to implement in practice, at least for the sorts of problems studied in first-year calculus.

We eschew the entire business, and go straight to the “just ask *Maple*” method. We extend our previous example to a nonhomogeneous problem where $G(x) = \sin x$.

$$\left[\begin{array}{l} > \text{dsolve}\left(D^{(2)}(y)(x) - D(y)(x) - y(x) = \sin(x)\right) \\ & y(x) = e^{\frac{1}{2}x(1+\sqrt{5})} _C2 + e^{-\frac{1}{2}x(1+\sqrt{5})} _C1 + \frac{1}{5} \cos(x) - \frac{2}{5} \sin(x) \end{array} \right]$$

We can clearly see our homogeneous solution there (even though the arbitrary constants have moved from the front to the back of each term), and an extra bit at the end which is, presumably, the particular solution. We have a closer look at that.

$$\left[\begin{array}{l} > y := x \rightarrow \frac{1}{5} \cos(x) - \frac{2}{5} \sin(x); \\ & D^{(2)}(y)(x) - D(y)(x) - y(x) \\ & \qquad \qquad \qquad y := \frac{1}{5} \cos(x) - \frac{2}{5} \sin(x) \\ & \qquad \qquad \qquad \sin(x) \end{array} \right]$$

So it is clear that $\frac{1}{5} \cos x - \frac{2}{5} \sin x$ is a solution to the original nonhomogeneous problem. It should be clear, using the fact that

$$\frac{d}{dx}(f + g) = \frac{d}{dx}(f) + \frac{d}{dx}(g)$$

that the entire function returned by *Maple* is also a solution to the equation as well.

One might well ask why, when we have a perfectly good and easy-to-write solution such as $\frac{1}{5} \cos x - \frac{2}{5} \sin x$ would we ever want to bother with adding in the extra mess of the solution to the complementary equation as well. The answer is that $y_c + y_p$ is the *general* solution to the equation. That is to say that any and every solution to the equation can be written in that form. Indeed, the particular solution, above, was the same as the general solution with $_C1 = _C2 = 0$. The proof of this fact is quite elementary, and can be found in [12] if desired.

2.2.5 Parametric Equations, Alternative co-ordinates, and Other Esoteric Plotting Fun


Recall that when a graph is plotted, what we are seeing is a graphical representation of pairs of points that satisfy some relationship. The parabola, for example, is the collection of all points (x, y) in $\mathbb{R} \times \mathbb{R}$ where $y = x^2$.

We may also plot functions from parametric equations, where a parameter, t say, varies, and the points in the plot are of the form $(x, y) = (x(t), y(t))$. There is not necessarily a relationship between the x and y co-ordinates in a parametric equation, apart from the fact that they share the same t -value. Of course, any function $f(x)$ may be turned into a parametric equation $(x, y) = (t, f(t))$.

Maple's **plot** function will allow us to plot parametric equations, if we so wish. To do so, we must pass a three-element list as the first argument to the **plot** function, instead of the usual expression. This list must be in the form of $[x(t), y(t), t = a..b]$ where $a..b$ is a range, t is any valid variable name, and $x(t)$ and $y(t)$ are arbitrary expressions involving that variable. Note that *Maple* will automatically scale the horizontal and vertical axes to fit the plot, based on the values of $x(t)$ and $y(t)$ as t varies through its range.

However, one should be careful; the axis ranges are independent of the parameter range, and all three may be modified separately. In the case of our parabola, there is a very direct relationship between the parameter t , and the horizontal and vertical ranges. This need not be so.

For example, to plot our parabola using parametric equations, we would input the following.




To see the independence of the axis and parameter variables, recall the parametric equations of a circle centered at an arbitrary point, (x_0, y_0) say, are

$$(x, y) = (x_0, y_0) + (r \cos \theta, r \sin \theta) = (x_0 + r \cos \theta, y_0 + r \sin \theta)$$

This is easiest to see by treating these parametric equations as vector equations, and recalling that the parametric equations of a circle centered at the origin are $(x, y) = (r \cos \theta, r \sin \theta)$.

We plot a circle, centered at $(1, 2)$, with radius 3. We should expect the horizontal range to be -2.4 and the vertical range to be -1.5 .




In order to demonstrate the independent nature of the axis and parameter ranges, we specifically instructed *Maple* to plot a larger axis range. We can clearly see the circle neatly within the ranges as defined by the parameter, and yet *Maple* has happily plotted with the extra range we requested.


In a lot of (if not most) cases separately modifying the horizontal, vertical, and parameter ranges will not be necessary. Nonetheless, it is worth being aware of the fact that they may be independently specified.

The circle example demonstrates a very useful feature of parametric plots, which is that they may be used to plot curves that are not the result of functions. The circle is clearly not a function, as it violates the vertical line test. Recall that a function always associates a single value in the range with each single value in the domain. This is not the case with a circle; we cannot assign a function $y = f(x)$ that will produce all the points in a circle. We can, of course, simply plot $y = \pm\sqrt{1-x^2}$ and display them together, but doing this is often neither easy nor even possible.

To further illustrate this advantage of parametric equations, we plot a spiral using parametric equations. We use the parametric equations $(x, y) = (t \cos t, t \sin t)$. This varies from the circle in that the radius is no longer fixed. If we think of the points (x, y) as vectors, then each point on the line is a vector of length t and angle t . As t increases, then the angle will cycle, but the length will continue increasing. We plot four full revolutions of this spiral.



Returning to our circle plots, the reader may well recall that although there is no explicit function for a circle, there most certainly is an equation that gives an implicit function for the circle. That equation is, of course, $(x - x_0)^2 + (y - y_0)^2 = r^2$ where (x_0, y_0) is the center of the circle and r is the radius. One may well wonder if *Maple* can plot such implicit equations. The answer is that yes it can, although we need to use a special function in the **plots** package, which goes by the name of **implicitplot**.



Notice that even though we asked for the plot to be in the extra range, just as we did with the parametric plot above, the **implicitplot** function plotted the circle to its extremities, and no more. It would seem that when evaluating implicit equations, *Maple* evaluates all the pairs (x, y) within the range that satisfy the equation, and then works out the required scale for the axes.


So far we have considered only Cartesian co-ordinates (of the form (x, y) in relation to two axes) when plotting. Another common co-ordinate system is the so-called *polar* co-ordinates, which are co-ordinates of the form (r, θ) where θ is the angle, and r is the distance traveled in that direction. The polar co-ordinates $(\sqrt{2}, \frac{1}{4}\pi)$, for example, correspond to the Cartesian co-ordinates $(1, 1)$. We may freely convert between polar and Cartesian co-ordinates with the identities $r = \sqrt{x^2 + y^2}$, $x = r \cos \theta$, and $y = r \sin \theta$. These identities may easily be confirmed by drawing up a trigonometric triangle.

Maple will happily plot polar equations (i.e., equations given in terms of polar co-ordinates) for us either on a regular Cartesian pair of axes, or on a special background more suited to the polar co-ordinates. The latter requires a special function in the **plots** package. Polar plots expect an expression for r to be a function of θ (just as regular plots expect an expression for y as a function of x). This should be unsurprising, given that polar equations are usually written as $r = r(\theta)$.

Let us start with a circle, as it is quite simple. A circle contains points that are equidistant from its center, hence r is the constant radius, and θ may take any value. So our polar equation is simply $r = C$ where C is a constant. Such a circle, however, will always be centered at the origin. Plotting, in polar co-ordinates, a circle not centered at the origin is trickier. See Exercise 10b.


Let us look at the spiral again. The spiral construction with parametric equations $(x, y) = (t \cos t, t \sin t)$ was constructed in a way that is very amenable to a polar equation. Recall that our t parameter varied as both an angle and a distance. This sounds very much like a polar equation. It should come as no surprise then that the polar equation of the spiral is simply $r = \theta$. To plot this, we simply pass the parameter `coords=polar` to the **plot** function.

```
> plot(theta, theta = 0..8 * Pi, coords = polar)
```



Alternatively we may use the **polarplot** function from the **plots** package. This has the advantage that it much better labels the r and θ parameters of the polar coordinate system, allowing us to much more easily read these values directly from the plot in much the same way we might read (x, y) value pairs from a plot on the plane.


```
> plots[polarplot](theta, theta = 0..8 * Pi)
```



For a more interesting and complicated example, we now look at a more complicated polar plot; $r = \sin(8\theta/5)$. This is clearly a cyclic equation, however the precise nature of the cycling needs some thought. It is clear that r will cycle every time θ changes through $5\pi/4$ radians. In order to cycle fully through all possible pairs of (r, θ) we need


to find integers a, b such that $a \cdot 2\pi = b \cdot 5\pi/4$ and the smallest such value for a is $a = 5$, corresponding to $b = 8$. We therefore plot this function for $0 \leq \theta \leq 10\pi$.

```
> plot(sin(8 * theta / 5), theta = 0..10 * Pi, coords = polar)
```



We may, using **implicitplot**, plot ranges of inequalities, with filled-in regions. We look at a couple of simple examples to close this section. The general approach is basically as one would expect; we replace the equation to be plotted with the inequality. For instance, if we wish to see the inequality $4x - 2y < 3$, we would try the following.


```
> plots[implicitplot](4x - 2y < 3, x = -1..1, y = -1..1)
```



The only catch is that in the above plot we have no idea on which side of the line the region we want lies, even though it is just a simple matter of substituting $x = y = 0$

into the inequality. In order to have *Maple* shade the relevant area, we need to use the **filledregions** option.


```
> plots[implicitplot](4x - 2y < 3, x = -1..1, y = -1..1, filledregions = true)
```



The option to change colors is a little different here as well. If we use the **color** option we will only change the color of the implicit curve itself, and not the regions. To color the regions we need to use **coloring**⁴ option, and we must use a list with this option, even for only one color. To demonstrate this, and to close off this section, we plot our familiar radius-3 circle, centered at (1, 2) with a black border and blue interior.

⁴ Note that, unlike other *Maple* options and keywords, the British spelling of “colouring” will not be recognised as an alternative spelling of **coloring**.

```
> plots[implicitplot]((x - 1)^2 + (y - 2)^2 ≤ 9, x = -3..5, y = -2..6, color =
    black, coloring = [blue], filledregions = true)
```




2.3 Multivariate Calculus

2.3.1 Three-Dimensional Plotting

The **plot** function, which is for two-dimensional plots, has a counterpart named **plot3d** which is rather unsurprisingly for three-dimensional plots. Ordinarily, **plot3d** will plot a function $z = f(x, y)$. To plot, for example, the paraboloid $z = x^2 + y^2$ we simply input the following.

```
> plot3d(x^2 + y^2, x = -1..1, y = -1..1)
```




Be aware that unlike the **plot** function, **plot3d** has no default values for the input range and so we must explicitly state ranges for both independent variables. Also be aware that 3-D plots default to having the z -axis pointing “up” (which is to say up with respect to the screen), unlike 2-D plots which have the y -axis pointing up. An advantage to these plots is that they may be rotated by dragging them with the mouse, which allows a better appreciation of the overall three-dimensional shape, even though we ultimately only have a two-dimensional projection on our computer screen. Of course,

such rotation will almost always change which, if any, axis is pointing up with respect to the screen.

Just as with the **plot** command and two-dimensional plots, we may parameterize surfaces and may have **plot3d** plot them for us. We do this by asking **plot3d** to plot a list of exactly three elements, instead of an expression. The three-element list is interpreted to be the parametric values for points $[x, y, z]$. Note that this is slightly different from the two-dimensional case, in that the range for the parameter(s) is not contained within the list.

```
> plot3d([v · cos(u), v · sin(u), v], u = 0..2 · Pi, v = -1..1)
```




Three-dimensional plots are usually—although not always—parameterized with two parameters. The trivial parameterization, for a function $f(x, y)$ is simply $[u, v, f(u, v)]$, and should look quite similar to the trivial parameterization of a function of one variable.

We look quickly at two alternative co-ordinate systems for three-dimensional surfaces. These are *cylindrical* and *spherical* co-ordinates. Both may be invoked with an appropriate **coords** option in the **plot3d** function.


Cylindrical co-ordinates are an extension of polar co-ordinates. Any point in 3-space may be located by specifying an angle on the xy -plane, and a distance from the origin at that angle (which is identical to polar co-ordinates), and finally by a height above (or below) the xy -plane. As such a point in cylindrical co-ordinates is of the form $[r, \theta, z]$, and as the name might suggest, it is quite easy to plot a cylinder using such co-ordinates.

A word of warning. *Maple*, by default, expects cylindrical co-ordinates to be expressed in the form of r as a function of θ and z . That is, $r = f(\theta, z)$. This is probably counterintuitive, as we were most probably expecting the form $z = f(r, \theta)$ just as for Cartesian co-ordinates, where we plot functions of the form $z = f(x, y)$. The easiest way around this problem is to plot cylindrical functions as parameterized plots of the form $[r, \theta, z]$, and is precisely what we do.

```
> plot3d([1, theta, z], theta = 0..2 · Pi, z = -1..1, coords = cylindrical)
```



```
> plot3d([r, theta, r], r = -1..1, theta = 0..2 * Pi, coords = cylindrical)
```




We show more cylindrical plots in Section 2.3.2.

Spherical co-ordinates are similar to cylindrical, differing only in the final co-ordinate. As with cylindrical co-ordinates, we have an r, θ pair that locates a point on the xy -plane. We then rotate that point *vertically* by an angle $\phi \in [0.. \pi]$ where the angle is measured against the z -axis, with 0 pointing upwards, and π pointing downward. Thus a point in 3-space in spherical co-ordinates has the form $[r, \theta, \phi]$. Our r co-ordinate in this system will always be the distance of a point from the origin (note that this is not the case with cylindrical co-ordinates). As the name might suggest, this co-ordinate system is very well suited to locating points on a sphere. Indeed, anybody familiar with longitude and latitude on maps of earth will have seen this concept before.


As with cylindrical co-ordinates, it is usually easier to plot spherical co-ordinate plots as parametric plots, as *Maple's* default is to expect r to be as a function of θ and ϕ .

```
> plot3d([1, theta, phi], theta = 0..2 * Pi, phi = -0..Pi coords = spherical)
```



We may also plot this sphere using the **implicitplot3d** function (from the **plots** package), which is a three-dimensional analogue of the **implicitplot** function from Section 2.2.5.

```
> plots[implicitplot3d](x^2 + y^2 + z^2 = 1, x = -1..1, y = -1..1, z = -1..1)
```




The interested reader is encouraged to check out the *Maple* help files on co-ordinates. See **?coords**, **?plot[coords]**, and **?plot3d[coords]**.

2.3.2 Surfaces and Volumes of Rotation

In general, calculating volumes is usually a job for iterated integrals (see Section 2.3.4). However, volumes of solids of revolution may be calculated with a single integral. A solid of revolution is produced by rotating a curve in the plane about a line. Usually, but not always, one of the axes is chosen. As an example, let us consider the sine curve between 0 and π .

```
> plot(sin(x), x = 0..Pi, scaling = constrained)
```




If we rotate the above sine curve about the x -axis we can imagine a sort of symmetrical teardrop shape. In fact, we can do better than imagine. We can have *Maple* draw us a picture. In order to plot our rotated sine curve, however, it is simplest to use a parametric equation. To this end, notice that at any point $x \in [0, \pi]$, our rotated surface will have a cross-section, parallel to the yz -plane which is a circle of radius $\sin(x)$. We may, therefore, parameterize the surface of the rotated surface by

$$[x, y, z] = [t, \sin(t) \sin(\theta), \sin(t) \cos(\theta)]$$

where $t \in [0, \pi]$ and $\theta \in [0, 2\pi)$.

This we may now plot.

```
> plot3d([t, sin(t) * sin(theta), sin(t) * cos(theta)], t = 0..Pi, theta = 0..2 * Pi,
        scaling = constrained)
```



The parameterization of this surface gives us a hint as to how to use integration to calculate the volume. We think of the volume as an infinite number of infinitely small disks (otherwise known as “circles”), and add up the area of each circle over an interval, the interval being $[0, \pi]$ in this case. The accumulated area is the volume. This is, incidentally, identical to a Riemann sum where we (more or less) add up the height of an infinite number of infinitely small boxes (otherwise known as “lines”) and accumulate these heights over an interval to obtain an area. The disks we are calculating are perpendicular to the axis of rotation.

We know that the area of any particular disk is πr^2 , and because our radius is $\sin(x)$, each disk in our particular example will have area $A(x) = \pi \sin(x)^2$. So the area,

remembering we have only rotated the portion of the sine curve between 0 and π , is

$$\int_0^\pi A(x) dx = \int_0^\pi \pi \sin(x)^2 dx = \pi \int_0^\pi \sin(x)^2 dx$$

This we may now calculate in *Maple* or manually, as we see fit.

```
> Pi · int(sin(x)^2, x = 0..Pi)
```

$$\frac{1}{2} \pi^2$$

In general, the volume of a solid produced by rotating a function $f(x)$ inside an interval $[a, b]$ around the x -axis is given by the integral


$$\pi \int_a^b f(x)^2 dx$$

If we wish to rotate a function $f(x)$ around the y -axis, then we need to rewrite the function as a function of y instead. Note that this is more complicated than just replacing every x in the function with a y . For example, let us rotate the parabola $y = x^2$ around the y -axis for $x \in [0, 2]$.

We start by plotting what we want to see. First, however, remember that with a three-dimensional plot it is the z -axis which is pointing upwards, but for a two-dimensional plot it is the y -axis that is pointing upwards. This is easily fixed by just renaming the y -axis to be the z -axis instead, giving us our “new” function of $z = x^2$, being rotated around the z -axis. For the remainder of this section, and any time we discuss volumes of revolution, we consider the 2-D plots to be in the xz -plane.


Once this is done, we notice that when rotating a curve around the z -axis, we have a natural cylindrical co-ordinate representation, r representing the distance from the origin (between 0 and 2), θ as the angle, and $z = r^2$ as the height above that point.

```
> plot3d([r, theta, r^2], r = 0..2, theta = 0..2 · Pi, coords = cylindrical)
```




Straight away we have an issue (although it is not immediately obvious). It is not clear from this plot whether we mean the volume between the paraboloid and the z -axis, or the volume “below” the paraboloid, down to the xy -plane. In fact we meant the latter, which is troublesome to visualize on that plot.

We remedy matters here. It is useful to be explicit about precisely which area we are rotating around the z -axis. In this case, we wish to take the volume under the curve $z = x^2$, but above the x -axis, and between the values $x = 0$ and $x = 2$. By plotting these bounds in two dimensions we should make it clearer exactly which area will be rotated.




We can't see the line that forms the lower bound, because the x -axis is obscuring it, nonetheless it is quite clear now exactly which area is being rotated. We plot the volume of revolution now, and again—just as with the 2-D plot above—we include the bounds as well. The line where $x = 2$ becomes a cylinder when rotated, and the line at the bottom of the plot becomes a disk. We are also cunning and leave a part of the solid open, so that we can see inside for a better impression of the solid in question.

```
> plot3d([[x, theta, x^2], [2, theta, x^2], [x, theta, 0]], x = 0..2, theta =  
0.. $\frac{5 \cdot \text{Pi}}{3}$ , coords = cylindrical)
```



Now in order to use our integral, above, we need to be rotating around the same axis as the independent variable of our function. Here, however, we are rotating a function of x around the z -axis (still using the z -axis as the upward pointing one). We need to rewrite our function as $x = f(z)$. Rearranging $z = x^2$ in this manner produces $x = \sqrt{z}$. We also notice that $x = 0 \implies z = 0$ and $x = 2 \implies z = 4$, so we may equivalently consider our rotation as rotating the function $x = \sqrt{z}$ for $z \in [0, 4]$ around the z -axis. However, it is the area of the function above $x = \sqrt{z}$ that we are rotating around the z -axis. The upper bound for this function is the line $x = 2$.



So, what we actually want is the area between the two curves $x = 2$ and $x = \sqrt{z}$ for $0 \leq z \leq 4$ to be rotated. That's easy. Observe that $x = 2$ is always larger than $x = \sqrt{z}$ on the interval in question. For the area itself, we could happily calculate the integral of the larger minus the integral of the smaller. We may take the same approach with the volume integral. If we calculate the volume of the cylinder we obtain by rotating $x = 2$ around the z -axis, and subtract from that the volume of the paraboloid (obtained by rotating the volume under the function $x = \sqrt{z}$) then we have calculated the precise area we wanted. This is demonstrated in Figure 2.2.




Fig. 2.2 Cylinder with paraboloid removed.

In short, we wish to calculate


$$\pi \int_0^4 2^2 dz - \pi \int_0^4 \sqrt{z}^2 dz = \pi \int_0^4 4 - z dz$$

```
[ > Pi · int(4 - z, z = 0..4)
```

8π


Let us return now to our sine function, plotted between $x = 0$ and $x = \pi$. We now rotate this around the z -axis (treating z as the axis pointing “up” again, as we did with the last example). We start as we have each other time, with plotting the surface to see with what we’re dealing. We add a **scaling** parameter to make the plot clearer in this case.

```
[ > plot3d([x · sin(theta), x · cos(theta), sin(x)], x = 0..Pi, theta = 0..2 · Pi,
    scaling = constrained)
```



Now we have a problem here, inasmuch as there’s no particularly easy or, at least, obvious way to rewrite $z = \sin(x)$ as a function of z . The z -interval is clearly $[0, 1]$, however, for any given value of z there are two values of x . This is easily seen with a plot.

```
[ > plot([sin(x), 0.9], x = 0..Pi, z, color = [red, blue], scaling = constrained)
```



The line $z = 0.9$ clearly cuts the sine function in two places.

Now, we may try and express the solid of rotation as an area between two functions, $f(z), g(z)$ and calculate the integral accordingly. However finding these two functions is tedious and time consuming. Instead, we use a slightly different integral to calculate the area. Our previous method used disks that were perpendicular to the axis of rotation. This time we use a method known as “shells”. To do this, we approximate the area as infinitely many cylinders, centered at the origin, and with radius x and height $\sin(x)$. The surface area of each cylinder is the circumference multiplied by the height. As such, the area function is $A(x) = 2\pi x \sin(x)$, and our volume may now be calculated as

$$\int_0^\pi A(x) dx = \int_0^\pi 2\pi x \sin(x) dx = 2\pi \int_0^\pi x \sin(x) dx$$

```
[ > 2 · Pi · int(x · sin(x), x = 0..Pi)
```

$2\pi^2$

In general if we have a function that is an n th partial derivative, which we then take yet another partial derivative of, then we have the following scenario.

$$(f_{x_1, \dots, x_n})_{x_{n+1}} = f_{x_1, \dots, x_{n+1}} \quad \text{or} \quad \frac{\partial}{\partial x_{n+1}} \left(\frac{\partial^n}{\partial x_n \cdots \partial x_1} \right) = \frac{\partial^{n+1}}{\partial x_{n+1} \cdots \partial x_1}$$

Note that the above also demonstrates why the ∂ notation has the variables written backwards.

Like standard derivatives, partial derivatives are defined in terms of the limit of the slopes of a series of lines between the point at which we wish to calculate the derivative, and another point near to it, as follows.

$$f_x(x, y) := \lim_{h \rightarrow 0} \frac{f(x+h, y) - f(x, y)}{h} \quad \text{or} \quad f_y(x, y) := \lim_{h \rightarrow 0} \frac{f(x, y+h) - f(x, y)}{h}$$

Maple may perform partial derivatives. Indeed, the **diff** function we have already used for regular differentiation will happily perform partial differentiation. For a first partial derivative, we tell **diff** with respect to which variable we want to take the derivative.

```
> p := sum(sum(x^i * y^j, j = 0..2), i = 0..2);
diff(p, x); diff(p, y)
      2 + 2 y + 2 y^2
      1 + 2 y + 2 x + 4 x y
      1 + 2 y + 2 x + 4 x y
      2 + 2 x + 2 x^2
```

And for second and later partial derivatives, we simply list the derivatives in the order they are to be taken. Note that this is exactly what we did for single variable derivatives, only there was only one variable in that case. So $\partial^2 f / \partial x^2$ was calculated with `diff(f(x), x, x)` which told *Maple* to take the first derivative with respect to x and then the second derivative with respect to x . Multivariable derivatives are different only in that we may take the derivative with respect to a different variable at each step.

```
> diff(p, x, x); diff(p, x, y);
diff(p, y, x); diff(p, y, y)
      2 + 2 y + 2 y^2
      1 + 2 y + 2 x + 4 x y
      1 + 2 y + 2 x + 4 x y
      2 + 2 x + 2 x^2
```

Of course, the inert form of the function still works, as do—as suggested above—higher-order derivatives.

```
> Diff(p, x, x, y) = diff(p, x, x, y)
      \partial^3
      \partial y \partial x^2 (1 + y + y^2 + x + x y + x y^2 + x^2 + x^2 y + x^2 y^2) = 2 + 4 y
```

The **D** function may also be used. For a function of multiple variables, to compute the first partial derivative with respect to the first variable, we use the subscript 1 with the **D**. To compute the first partial derivative with respect to the second variable, we use a subscript of 2, and so on. For example,

$$\left[\begin{array}{l} > f := (x, y) \rightarrow x^2 - y^2 + x \cdot y; D_1(f), D_2(f) \\ & f := (x, y) \rightarrow x^2 - y^2 + xy \\ & (x, y) \rightarrow 2x + y \\ & (x, y) \rightarrow -2y + x \end{array} \right]$$

To compute second partial derivatives, we simply subscript **D** with a sequence of numbers describing the variable numbers, in the order that the derivatives are taken. The following examples may be quickly checked by hand.

$$\left[\begin{array}{l} > Diff(f(x, y), x\$2) = D_{1,1}(f)(x, y) \\ & Diff(f(x, y), y, x) = D_{2,1}(f)(x, y); \\ & \frac{\partial^2}{\partial x^2} (x^2 - y^2 + xy) = 2 \\ & \frac{\partial^2}{\partial x \partial y} (x^2 - y^2 + xy) = 1 \end{array} \right]$$

Note that although this is different from the method we used to take derivatives of single variable functions with the **D** command, this method also works for single variable functions.

Recall that for a function with continuous second partial derivatives, then the second partial derivatives, $f_{y,x}$ and $f_{x,y}$ will be equal. This is Clairaut's theorem (see [12]). More precisely:

Theorem 1 (Clairaut's). *Let f be a function defined on a disk $D \subset \mathbb{R}^2$, such that all the second partial derivatives are continuous on D . Then $f_{y,x}(a, b) = f_{x,y}(a, b)$ for every $(a, b) \in D$.*

It follows, of course, that if the function f is defined for all of $\mathbb{R}^2$ and its partial derivatives are continuous on all of $\mathbb{R}^2$ then it will certainly be the case that $f_{y,x} = f_{x,y}$.

We have seen an example of this above, with the second partial derivatives of our polynomial p . A quick check of our first partial derivatives of our function f shows that the second partial derivatives $f_{x,y}$ and $f_{y,x}$ are clearly equal. Let's look at another example.

$$\left[\begin{array}{l} > f := \sin(x^2 + y^2); diff(f, x, y) = diff(f, y, x); \\ & f := \sin(x^2 + y^2) \\ & -4 \sin(x^2 + y^2) yx = -4 \sin(x^2 + y^2) yx \end{array} \right]$$

And this even extends to higher partial derivatives.

$$\left[\begin{array}{l} > diff(f, x, x, y); diff(f, x, y, x); diff(f, y, x, x) \\ & -8 \cos(x^2 + y^2) yx^2 - 4 \sin(x^2 + y^2) y \\ & -8 \cos(x^2 + y^2) yx^2 - 4 \sin(x^2 + y^2) y \\ & -8 \cos(x^2 + y^2) yx^2 - 4 \sin(x^2 + y^2) y \end{array} \right]$$

For an example where Clairaut's theorem does not hold, see Exercise 14.

The partial derivatives, just as regular derivatives, allow calculation of a line that is tangent to a surface, $f(x, y)$ say. However, there are potentially many such tangent lines in a three-dimensional space. As such, the partial derivatives f_x and f_y give the slope of a tangent line parallel to the x - or y -axes, respectively.

Lines in three-dimensional space are tricky, but may be plotted with a parametric equation. In the case of our directional derivatives, we know a point on the line, the direction each line is traveling, and the rate at which the height of each line changes (the slope). In short, we have all the information we need to plot the tangent lines.

For a surface $f(x, y)$ we may use the following parameterization of the tangent lines.

$$\begin{aligned} [x_0 + u, y_0, f(x_0, y_0) + u \cdot f_x(x_0, y_0)] & \text{ in the } x \text{ direction} \\ [x_0, y_0 + v, f(x_0, y_0) + v \cdot f_y(x_0, y_0)] & \text{ in the } y \text{ direction} \end{aligned}$$


and, of course, the obvious parameterization of the surface itself as $[u, v, f(u, v)]$.

```

> tl_x := proc(f :: procedure)
    (x0, y0) → [x0 + u, y0, f(x0, y0) + u · D1(f)(x0, y0)]
end :
tl_y := proc(f :: procedure)
    (x0, y0) → [x0, y0 + v, f(x0, y0) + v · D2(f)(x0, y0)]
end :

> f := (x, y) → x^2 + y^2 :
plot3d([ [u, v, f(u, v)], tl_x(f)(0, 1), tl_y(f)(0, 1)], u = -1..1, v = -1..1,
    axes = framed)

```



The directional derivatives allow us to calculate the *tangent plane* to a surface in 3-D. Given a function, $f(x, y)$ say, with continuous derivatives the equation of the tangent plane to the surface of f at a point (x_0, y_0, z_0) is

$$z - z_0 = \frac{\partial f}{\partial x}(x_0, y_0) \cdot (x - x_0) + \frac{\partial f}{\partial y}(x_0, y_0) \cdot (y - y_0)$$

which may be rewritten as

$$\begin{aligned} z &= \frac{\partial f}{\partial x}(x_0, y_0) \cdot (x - x_0) + \frac{\partial f}{\partial y}(x_0, y_0) \cdot (y - y_0) + z_0 \\ &= \frac{\partial f}{\partial x}(x_0, y_0) \cdot (x - x_0) + \frac{\partial f}{\partial y}(x_0, y_0) \cdot (y - y_0) + f(x_0, y_0) \end{aligned}$$

because if (x_0, y_0, z_0) is a point on the surface of $f(x, y)$, then it must be the case that $z_0 = f(x_0, y_0)$.

We may (and, indeed, do) explore this in *Maple*. First we create a procedure that returns an arrow-notation function for the tangent plane equation.

```

> tp := proc(f :: procedure)
    (x0, y0) → f(x0, y0) + D1(f)(x0, y0) · (x - x0) + D2(f)(x0, y0) · (y - y0)
end :


```

Now we have a function tp which, when given a procedure as input, will return a new function that calculates the tangent plane at a given point. Because the tp creates a function, we expect to be able to use the expression $tp(f)(a, b)$ to calculate the tangent plane to the function f at the point (a, b) .


```
> f := (x, y) → x2 + y2; tp(f)(a, b)
      f := (x, y) → x2 + y2
      a2 + b2 + 2 a (x - a) + 2 b (y - b)
```

Well, that's all well and good, but it's about time we produced some plots.

```
> plot3d([f(x, y), tp(f)(0, 0)], x = -2..2, y = -2..2);
```



```
> plot3d([f(x, y), tp(f)(1, 1)], x = -2..2, y = -2..2);
```



We may calculate tangent lines (and so instantaneous rate of change) at a point in both the x - and y -directions as we wish. Furthermore, we may use this information to calculate the tangent plane at the same point. It should follow, therefore, that we ought to be able to calculate the slope of a tangent line in any direction, and also that it should lie on the tangent plane we have already calculated.

To do this, we need a vector of unit length pointing in the direction we wish to calculate the slope. Call this vector $u = (a, b)$. Then we may calculate the directional derivative in the direction of u at any point (x, y)

$$D_u f(x, y) := f_x(x, y) \cdot a + f_y(x, y) \cdot b$$

We show that this vector is on the tangent plane. If we ignore the z -axis for the moment, we can parameterize the line from a point (x_0, y_0) in the direction of u as $[x_0 + ta, y_0 + tb]$ (think of the vector equation $(x_0, y_0) + t \cdot (a, b)$). Getting back to thinking three-dimensionally, we know that this line is climbing at the rate of $D_u f(x_0, y_0)$ in the z axis for each unit moved in the direction of u , which is supposed to be of unit length anyway.

This leads us to the parameterization of the line as

$$[x_0 + t \cdot a, y_0 + t \cdot b, f(x_0, y_0) + t \cdot D_u f(x_0, y_0)]$$

and so if a point on this line is on the plane, which we should recall has equation

$$z = f(x_0, y_0) + f_x(x_0, y_0) \cdot (x - x_0) + f_y(x_0, y_0) \cdot (y - y_0)$$

then it must be the case that every point on the line as parameterized above, satisfies the equation. Over to *Maple* now.

```

> f := ' f ' :
   tp := (x0, y0) -> f(x0, y0) + D1(f)(x0, y0) * (x - x0) + D2(f)(x0, y0)
   . (y - y0);
   du := (x0, y0) -> D1(f)(x0, y0) * a + D2(f)(x0, y0) * b;
tp := (x0, y0) -> f(x0, y0) + D1(f)(x0, y0) (x - x0) + D2(f)(x0, y0) (y - y0)
du := (x0, y0) -> D1(f)(x0, y0) a + D2(f)(x0, y0) b

> subs({x = x0 - t * a, y = y0 - t * b}, tp(x0, y0))
   f(x0, y0) + t * (du(x0, y0));
   f(x0, y0) + D1(f)(x0, y0) ta + D2(f)(x0, y0) tb
   f(x0, y0) + t (D1(f)(x0, y0) a + D2(f)(x0, y0) b)

> simplify(% - %%)
0

```

And there we have it. Notice that we needed to reset the f variable in order to maintain proper generality. We could probably have done without the final subtraction, inasmuch as it was clear that the two calculated expressions were the same, but it never hurts to check and we didn't have to go out of our way to do so. In any event, it is clear that the directional derivative in the direction of a unit vector u will give us the slope of a tangent line that lies on the tangent plane and travels parallel to the direction of u .

2.3.4 Double Integrals

Just as the space under a curve, but above the x -axis, may be calculated as an area via integration, so too can the space under a surface (i.e., a function $z = f(x, y)$), and above the xy -plane, may be calculated as a volume using integration.

The same basic approach applies. Given a range for x and y we have a rectangle that is a subsection of the xy -plane. We partition the x and y ranges, resulting in the area under the surface being partitioned into rectangles. We then create rectangular prisms

by choosing a point inside every rectangle, and setting the height of that rectangular prism to be the height of the function evaluated at that point. This should sound awfully similar to the method of approximating the area under a curve with rectangles.

If we let A_{ij} be the area of the (i, j) th subrectangle (i.e., the rectangle that is at the intersection of the i th x -partition and j th y -partition), and (x_{ij}^*, y_{ij}^*) be a point inside that subrectangle, then the area of the resulting rectangular prism is $f(x_{ij}^*, y_{ij}^*)A_{ij}$. The volume underneath the surface (and above the xy -plane) may be approximated by adding up the volumes of all the rectangular prisms.


This is well demonstrated in *Maple* using the **ApproximateInt** command from the **Student[MultivariateCalculus]** package. Note that in the *Maple* worksheet itself, the following two commands produce animations, and not just static pictures. Such animation would be long and tedious to produce by hand (although quite possible with the **animate** command from the **plots** package). Note that the text above the images is automatically generated and packaged part of the output of **ApproximateInt**.

We may, as it happens, use a more interactive manner of exploration. *Maple* provides a number of interactive demonstration screens, refereed to as *tutors*. In particular is the **ApproximateIntTutor**, which may also be used to produce the following animations. The tutor has the advantage of being interactive, allowing experimentation with quickly seen results.

The tutors are best used for exploration. Unfortunately due to their interactive nature, they are troublesome to explain in print, and so fall out of the scope of this book. They may be accessed through the *Tools* menu, which has a *Tutors* submenu, and cover many areas of university mathematics appropriate to students. The reader is encouraged to explore these for themselves (see also Exercise 1).

```
> Student[MultivariateCalculus][ApproximateInt](x^2 + y^2, x = -2..2, y = -2..2,
method = midpoint, coordinates = cartesian, partition = [25, 25],
output = animation)
```


An animation of approximations to the integral of
 $f(x, y) = x^2 + y^2$
over the region $[-2 \dots 2, -2 \dots 2]$
using a midpoint Riemann sum
Actual value: 42.667



The above is using midpoints as the sample points (x_{ij}^*, y_{ij}^*) . Random points work just as well, as it happens (and, incidentally, also work for Riemann sums of single variable functions).

```
> Student[MultivariateCalculus][ApproximateInt](x^2 + y^2, x = -2..2, y = -2..2,
  method = random, coordinates = cartesian, partition = [25, 25],
  output = animation)
```

An animation of approximations to the integral of
 $f(x, y) = x^2 + y^2$
 over the region $[-2 \dots 2, -2 \dots 2]$
 using a random Riemann sum
 Actual value: 42.667



We obtain the volume under the curve by taking progressively more and more rectangular prisms (which are, in turn, smaller and smaller). As such the volume under the surface may be approximated by the double sum

$$\sum_{i=1}^n \sum_{j=1}^m f(x_{ij}^*, y_{ij}^*) A_{ij}$$

and the volume as the limit of this sum as both n and m approach ∞ .

$$V = \lim_{n, m \rightarrow \infty} \sum_{i=1}^n \sum_{j=1}^m f(x_{ij}^*, y_{ij}^*) A_{ij}$$

The question now is, how do we turn this into an integral? The key lies in recognizing that the two sums (above) may be taken to infinity independently. In integration terms, fix y as a constant, and calculate the integral of the function as if it were just a function of the single variable x . The result will be a function of y that tells us the *area* under the surface for any given y value. Integrating this function with respect to y will give us the volume. This is, in fact, very similar to our volumes of revolution from Section 2.3.2. If we think of this as accumulation, we accumulate an infinite amount of areas to obtain a volume, just as we did with the disks or cylinders of the solids of revolution.

To make this more rigorous, let $f(x, y)$ be a function of two variables. We wish to find the volume under the surface of f for $x \in [x_1, x_2]$ and $y \in [y_1, y_2]$. In other words we want to integrate over the rectangle $[x_1, x_2] \times [y_1, y_2]$. Furthermore, suppose that $f(x, y)$ is *continuous* over that rectangle. The integral $\int_{x_1}^{x_2} f(x, y) dx$ is used to denote integrating with respect to x while holding y fixed. Similarly $\int_{y_1}^{y_2} f(x, y) dy$ denotes integrating with respect to y while holding x to be fixed. These are called *partial integrals* (compare to partial derivatives).

Then the volume can be calculated as

$$V = \int_{y_1}^{y_2} \left(\int_{x_1}^{x_2} f(x, y) dx \right) dy$$

which, for simplicity's sake is usually just written without the bracketing, as it is understood that the innermost integral needs to be performed first, before the outermost integral may be performed

$$V = \int_{y_1}^{y_2} \int_{x_1}^{x_2} f(x, y) dx dy$$

Let's look at this in *Maple* using our favorite paraboloid, over the rectangle $[-2, 2] \times [-2, 2]$.

$$\left[\begin{array}{l} > f := x^2 + y^2; \\ & \text{Int}(f, x = -2..2) = \text{int}(f, x = -2..2) \\ & \qquad \qquad \qquad f := x^2 + y^2 \\ & \qquad \qquad \qquad \int_{-2}^2 (x^2 + y^2) dx = \frac{16}{3} + 4y^2 \\ > \text{Int}(\text{rhs}(\%), y = -2..2) = \text{int}(\text{rhs}(\%), y = -2..2) \\ & \qquad \qquad \qquad \int_{-2}^2 \left(\frac{16}{3} + 4y^2 \right) dy = \frac{128}{3} \end{array} \right]$$

We see from the above that the first integration did indeed leave us with a function of y . We also see that, apparently, the volume is $128/3$ cubic units. We should hope, given the double sum definition, and the basic idea that we are calculating a volume, that if the order of integration is reversed, we should obtain the same answer. As it happens, we do indeed.

$$\left[\begin{array}{l} > \text{Int}(f, y = -2..2) = \text{int}(f, y = -2..2) \\ & \qquad \qquad \qquad \int_{-2}^2 (x^2 + y^2) dy = 4x^2 + \frac{16}{3} \\ > \text{Int}(\text{rhs}(\%), x = -2..2) = \text{int}(\text{rhs}(\%), x = -2..2) \\ & \qquad \qquad \qquad \int_{-2}^2 \left(4x^2 + \frac{16}{3} \right) dx = \frac{128}{3} \end{array} \right]$$

This guarantee that the double integral will always give the same answer, no matter the order the integrals are performed in, is given by Fubini's theorem. This guarantee is dependent on the function f being continuous over the rectangle in question. In fact, it is even true sometimes when f is not continuous over the rectangle, but we do not concern ourselves with the generalization.

Maple is often capable, it should come as no surprise to find, of handling double (even multiple) integrals without the need to manually perform each single integral. This is achieved by simply providing the ranges for x and y in a list, in the order they are to be performed.

$$\begin{aligned}
 &> \text{Int}(f, [x = -2..2, y = -2..2]) = \text{int}(f, [x = -2..2, y = -2..2]) \\
 &\quad \text{Int}(f, [y = -2..2, x = -2..2]) = \text{int}(f, [y = -2..2, x = -2..2]) \\
 &\quad \int_{-2}^2 \int_{-2}^2 (x^2 + y^2) \, dx dy = \frac{128}{3} \\
 &\quad \int_{-2}^2 \int_{-2}^2 (x^2 + y^2) \, dy dx = \frac{128}{3}
 \end{aligned}$$


The reader may have noticed that the volume we get for this paraboloid is somewhat larger than the 8π cubic units of the paraboloid we obtained from revolving $z = x^2$ around the z -axis. This is explained by remembering that the volume of revolution was calculated with a circular base, whereas our paraboloid has a square base. The extra area under the corners (where, incidentally, the paraboloid realizes its largest values) explains the discrepancy.

To see this, we can adopt a couple of approaches. The first (and least satisfying) is to plot both surfaces together and see if we can see one completely contained in the other.

```

> P1 := plot3d(x^2 + y^2, x = -2..2, y = -2..2, color = blue) :
  P2 := plot3d([r, theta, r^2], r = 0..2, theta = 0..2 * Pi, coords = cylindrical,
    color = red) :
> plots[display]([P1, P2])

```



We can certainly see where the revolved paraboloid stops, and the other continues. However, a more compelling way of convincing ourselves of this would be to make a new function that is $x^2 + y^2$ as long as $\sqrt{x^2 + y^2} \leq 2$ and 0 otherwise. That is, we want

$$F(x, y) = \begin{cases} x^2 + y^2 & \text{if } \sqrt{x^2 + y^2} \leq 2 \\ 0 & \text{otherwise} \end{cases}$$

Integrating this function then should only give us the area of the paraboloid under the circle of radius 2 centered at the origin, and thus our 8π volume.

```

> F := (x, y) -> piecewise(sqrt(x^2 + y^2) <= 2, x^2 + y^2, 0);
  int(F(x, y), [x = -2..2, y = -2..2])
      F := (x, y) -> piecewise(sqrt(x^2 + y^2) <= 2, x^2 + y^2, 0)
                                     8 pi

```

Readers familiar with iterated integrals are able to verify this using integration techniques for type I or type II regions (see [12]). We content ourselves with the above.

Let us return to the idea of partial integration. We should expect the notions of integrals as antiderivatives to extend to partial integrals and partial differentiation. We explore this in *Maple*.

$$\left[\begin{array}{l} > \text{int}(f, x); \text{diff}(\%, x) \\ \\ \frac{1}{3}x^3 + y^2x \\ x^2 + y^2 \end{array} \right]$$

$$\left[\begin{array}{l} > \text{int}(f, y); \text{diff}(\%, y) \\ \\ x^2y + \frac{1}{3}y^3 \\ x^2 + y^2 \end{array} \right]$$

That certainly looks promising. Partial integration with respect to x or y is undone by partial differentiation with respect to x or y as appropriate. Also interesting is that if we follow the usual substitution done by hand in integration, we can probably obtain the definite integrals *Maple* calculated for us earlier.

$$\left[\begin{array}{l} > \text{int}(f, x); \text{subs}(x = 2, \%) - \text{subs}(x = -2, \%); \\ \\ \frac{1}{3}x^3 + y^2x \\ \frac{16}{3} + 4y^2 \end{array} \right]$$

$$\left[\begin{array}{l} > \text{int}(f, y); \text{subs}(y = 2, \%) - \text{subs}(y = -2, \%); \\ \\ x^2y + \frac{1}{3}y^3 \\ 4x^2 + \frac{16}{3} \end{array} \right]$$

That is, we have shown that

$$\left[\frac{1}{3}x^3 + y^2x \right]_{x=-2}^{x=2} = \frac{16}{3} + 4y^2 \quad \text{and} \quad \left[x^2y + \frac{1}{3}y^3 \right]_{y=-2}^{y=2} = 4x^2 + \frac{16}{3}$$

which is all as it should be, but is nonetheless nice to have verified.

Finally, we try a truly general function, and see if *Maple*'s partial differentiation and partial integration still undo each other.

$$\left[\begin{array}{l} > \text{Diff}(\text{Int}(g(x, y), x), x) = \text{diff}(\text{int}(g(x, y), x), x); \\ \text{Int}(\text{Diff}(g(x, y), x), x) = \text{int}(\text{diff}(g(x, y), x), x) \\ \\ \frac{\partial}{\partial x} \int g(x, y) dx = g(x, y) \\ \int \frac{\partial}{\partial x} g(x, y) dx = g(x, y) \end{array} \right]$$

$$\left[\begin{array}{l} > \text{Diff}(\text{Int}(g(x, y), y), y) = \text{diff}(\text{int}(g(x, y), y), y); \\ \text{Int}(\text{Diff}(g(x, y), y), y) = \text{int}(\text{diff}(g(x, y), y), y) \\ \\ \frac{\partial}{\partial y} \int g(x, y) dy = g(x, y) \\ \int \frac{\partial}{\partial y} g(x, y) dy = g(x, y) \end{array} \right]$$

Maple certainly seems to think that partial differentiation and partial integration are inverse procedures.

We have only just scraped the surface here with double integrals in particular, and multivariate calculus in general. The interested reader is encouraged to read [12] and/or

any other good calculus texts. Exploration and confirmation of such material should be possible with the tools used and discussed in this section.

2.4 Exercises

The exercises for this, and subsequent, chapters are less numerous and slightly longer in duration when compared to the exercises from Chapter 1. An effort has been made to keep the amount of work each question requires roughly the same for each question, and also for each question to be more or less self-contained. However, no guarantees to this effect are made.

1. *Maple* provides a group of packages all collected into a super-package named **Student**. The goal of these packages is to “assist with the teaching and learning of standard undergraduate mathematics.” In particular is the **Student[Calculus1]** package that deals with single variable calculus. Read the help information on this package (**?Student[Calculus1]**), giving particular (but not exclusive) attention to the *visualization* and *interactive* sections. Experiment with some of these commands and tutors.

Note: The tutors may alternatively be accessed directly through the *Tools* menu.

2. Plot the following functions. Make sure that, where possible, you plot all important information to the plot; turning points, zeroes, and so on.

a. $x^5 - 7x^4 - 162x^3 + 878x^2 + 3937x - 15015$

b. $\frac{\sin x}{x}$

c. $\frac{x}{\cos x - 1}$

d. $x^5 - 3x^4 + x^2 - x - 5$

The following plots produce slightly unexpected results. Plot the functions, and identify the unexpected behavior. Modify the **plot** parameters to produce a more correct plot. Also, have *Maple* plot the functions with identical scale for the vertical and horizontal axes.

e. $2 + \sin x$

f. $\sin(x)^2 + \cos(x)^2$

3. Evaluate the limits of the following functions for $x = \pm\infty$ as well as any undefined points. Produce appropriate plots to demonstrate these limits.

a. $\tanh x$

c. $\sin \frac{1}{x}$

b. $\frac{x^2 + 1}{x^2 - 1}$


d. $\cos \frac{1}{x}$

4. Find all critical points, maxima and minima, and inflection points for the function $y = x^4 + x$.
5. Using the substitution $u = \pi - x$ and the **IntegrationTools** package check the identity

$$\int_0^\pi x f(\sin x) dx = \frac{\pi}{2} \int_0^\pi f(\sin x) dx$$

Note: Recall that the **IntegrationTools** functions seem to work best with inert integrals.

6. A cone may be constructed by cutting a sector out of a circle, and then joining the two straight lines CA and CB (from the diagram to the right) which are created by the removal of arc. If the circle has radius R , then find a formula for the maximum volume that such a cone may have.



Hint: Try finding the circumference of the circle at the top of the cone.

Hint2: You will probably need to tell *Maple* some assumptions about your variables.

7. Evaluate the following integrals. In each case the answer is a combination (i.e., sums or products) of constants such as e , $\sqrt{2}$, $\sqrt{3}$, π , $\zeta(3)$, $\log 2$, and γ (Euler's gamma constant). At least one even involves $\log \pi$. If *Maple* cannot calculate the integral directly, evaluate it numerically and try to **identify** it.

a. $\int_0^{\frac{\pi}{2}} \frac{x^2}{\sin^2 x} dx$

c. $\int_0^{\frac{\pi}{2}} \frac{\arcsin\left(\frac{\sqrt{2}}{2} \sin x\right) \sin x}{\sqrt{4 - 2 \sin^2 x}} dx$

b. $\int_0^{\frac{\pi}{2}} \frac{x^4}{\sin^4 x} dx$

d. $\int_0^{\infty} \frac{\log x}{\cosh^2 x} dx$

The following two integrals arise from mathematical physics, but neither had a known closed form as of 2009. This may have changed.

e. $\int_0^1 \frac{\log\left(\sqrt{3+y^2}+1\right) - \log\left(\sqrt{3+y^2}-1\right)}{1+y^2} dy$

f. $\int_3^4 \frac{\operatorname{arcsec}(x)}{\sqrt{x^2 - 4x - 3}} dx$

8. Solve the following linear differential equations, verify the solution, and plot it for some values of the constant.

a. $xy' + y = x \cos x$ (for $x > 0$)

b. $y' + (\cos x)y = \cos x$

How do the solution curves change as the constant changes?

If you are feeling adventurous, try and plot the solutions using the **DEplot** function from the **DEtools** package.

9. The differential equation

$$x^2 y'' + xy' + (x^2 - \alpha^2)y = 0$$

is known as the Bessel equation. The solutions to this equation give rise to the so-called *Bessel functions of the first and second kind*, $J_\alpha(x)$ and $Y_\alpha(x)$, respectively.

- Solve the Bessel equation for the special cases of $\alpha = \frac{1}{2}$ and $\alpha = \frac{3}{2}$. Verify the solutions.
- Solve the Bessel equation for the general case. Verify the solution.
- Plot the Bessel functions J_α and Y_α for some values of α of your choosing.

The modified Bessel functions of the first and second kind— $I_\alpha(x)$ and $K_\alpha(x)$, respectively—are solutions of the modified Bessel equation,

$$x^2 y'' + xy' - (x^2 + \alpha^2)y = 0$$

- Solve the modified Bessel equation for the general case. Verify the solution.

- e. Plot the modified Bessel functions I_α and K_α for some values of α of your choosing.
10. a. Plot Pac-Man⁵ on a set of Cartesian axes. You need only produce the basic outline.
Hint: Use multiple polar co-ordinate plots and the **display** function.
 b. Find a polar equation for the circle of radius 3 with center $(1, 2)$. Plot the circle using this equation.
11. a. Find the volume obtained by rotating the area between by the curve $z = x^2$ and the z -axis for $x \in [0, 1]$ around the z -axis. Notice anything interesting?
 b. Plot and calculate the volumes obtained by rotating the following areas around the z -axis. In all cases the curve is $z = \log x$ with $x \in [0, 1]$.
 i. The area underneath the curve (between the curve and the x -axis)
 ii. The area between the curve and the z -axis.
 Check your answers by calculating both the disks and the shells method.
12. Find and classify the critical points of the following functions of two variables. Recall that critical points occur where $f_x(a, b) = f_y(a, b) = 0$. A critical point may be a maximum, a minimum, or a saddle point.

a. $f(x, y) := 3x^2y + y^3 - 3x^2 - 3y^2 + 2$. b. $f(x, y) := xye^{-x^2-y^2}$

13. Verify that the following functions are solutions to the partial differential equation $f_x + f_y = \sin(x) + \cos(y)$. A solution to a partial differential equation is a function whose partial derivatives satisfy the equation (compare to an ordinary differential equation).

a. $\sin(y) - \cos(x) + C \cdot (y - x)$ c. $\sin(y) - \cos(x) + \sqrt{y - x}$
 b. $\sin(y) - \cos(x) + (y - x)^n$ d. $\sin(y) - \cos(x) + e^{C \cdot (y - x)^n}$

What do you think the general solution might be?

14. Let f be the function defined below.

$$f(x, y) := \begin{cases} \frac{xy(x^2 - y^2)}{x^2 + y^2} & \text{if } (x, y) \neq (0, 0) \\ 0 & \text{if } (x, y) = (0, 0) \end{cases}$$

- a. Verify—visually or otherwise—that f is continuous at the origin.
 b. Show that $f_{xy}(0, 0) \neq f_{yx}(0, 0)$ and so Clairaut's theorem does not apply to this function.

Note: You need to use the limit definition on all functions to establish the value of their partial derivatives at the point $(0, 0)$.

15. Plot the following, and calculate their area using iterated integrals.
 a. The area underneath $z = x^5y^3e^{xy}$ for $0 \leq x \leq 1$ and $0 \leq y \leq 1$
 b. The area between $z = e^{-x^2}\cos(x^2 + y^2)$ and $z = 2 - x^2 - y^2$ for $|x| \leq 1$ and $|y| \leq 1$

The following iterated integrals are volumes underneath surfaces ($z = f(x, y)$) for points (x, y) that do not lie inside a rectangle. Calculate the integrals. Can you work out the bounding curves for the points (x, y) ?

⁵ Pac-Man is a video game character from the early 1980s. A quick Internet search should be all that is needed in order to know what the plot must look like.

c. $\int_1^2 \int_0^y x^2 y^2 \, dx \, dy$

d. $\int_0^2 \int_0^{\frac{x^2}{2}} \frac{x}{\sqrt{1+x^2+y^2}} \, dy \, dx$

2.5 Further Explorations

1. Finding limits.

- a. Let $a_0 = 0, a_1 = \frac{1}{2}$ and define

$$a_{n+1} := \frac{1 + a_n + a_{n-1}^3}{3}$$

Determine the limit as $n \rightarrow \infty$, and find out what happens when $a_1 = a$ is allowed to vary.

- b. Let $a_1 = 1$ and define

$$a_{n+1} := \frac{3 + 2a_n}{3 + a_n}$$

Determine the limit and find out what happens when $a_1 = a$ is allowed to vary.

The above two limits are easy enough to find and (depending on what you know) to prove.

- c. Let $a_1 \geq 1$ be given and determine the limit of the iteration

$$a_{n+1} := a_n - \frac{a_n}{\sqrt{1 + a_n^2}} + \sin(\theta)$$

for arbitrary θ .

2. A (*strict*) *mean* $M(a, b)$ is a continuous function of two positive numbers that calculates a number c lying between a and b (strictly between them as long as $a \neq b$). The arithmetic and geometric means are clearly such objects. A mean iteration takes two means, M and N say, and iterates by setting $a_0 = a, b_0 = b$ and

$$a_{n+1} := M(a_n, b_n), \quad b_{n+1} := N(a_n, b_n)$$

The limit of such a strict mean iteration always exists and is denoted by $M \otimes N(a, b)$. Identify the limits of the mean iteration defined by the means in the following.

- a. $M(a, b) := \frac{a+b}{2}, N(a, b) := \frac{2ab}{a+b}$
 b. $M(a, b) := \frac{a + \sqrt{ab}}{2}, N(a, b) := \frac{b + \sqrt{ab}}{2}$

3. Define

$$\text{sinc}(x) := \frac{\sin x}{x}$$

and explore the following integrals. Calculate them for all (natural) values of N from 1 to 8 at least (more if you wish), and measure the time each calculation takes to perform. Can you work out what is going on?

a. $\int_0^\infty \prod_{n=0}^N \operatorname{sinc}\left(\frac{x}{2n+1}\right) dx$

b. $\int_0^\infty \prod_{n=0}^N \operatorname{sinc}\left(\frac{x}{3n+2}\right) dx$

An Introduction to Modern Mathematical Computing
With Maple™

Borwein, J.; Skerritt, M.P.

2011, XVI, 216 p., Hardcover

ISBN: 978-1-4614-0121-6